
Nagyhatékonyságú logikai programozás

A tárgy témakörei

• Korlát-logikai programozás (CLP — Constraint Logic Programming)

• A Mercury „nagybani” logikai programozási nyelv

Információk a korlát-logikai programozásról

• „Sárga könyv”: Kim Mariott, Peter J. Stuckey, Programming with Constraints:
an Introduction, MIT Press 1998 (részletesebben lásd
http://www.cs.mu.oz.au/~pjs/book/book.html )

• „Az első alapkönyv”: Pascal Van Hentenryck: Constraint Satisfaction in Logic
Programming, MIT Press, 1989

• On-line Guide to Constraint Programming, by Roman Barták
(http://kti.ms.mff.cuni.cz/~bartak/constraints/ )

Információk a Mercury nyelvr ől

• Honlap:http://www.cs.mu.oz.au/research/mercury/

A CLP alapgondolata

A CLP(X ) séma

Prolog +
egy valamilyenX adattartományra és azon értelmezett korlátokra
(relációkra) vonatkozó „erős” következtetési mechanizmus.

Példák azX tartomány megválasztására

X = Q vagy R (a racionális vagy valós számok)
korlátok= lineáris egyenl̋oségek és egyenlőtlenségek
következtetési mechanizmus= Gauß elimináció és szimplex módszer

X = FD (egész számok Véges Tartománya, angolul FD — Finite Domain)
korlátok= különféle aritmetikai és kombinatorikus relációk
következtetési mechanizmus= MI CSP–módszerek (CSP = Korlát-Kielégítési
Probléma)

X = B (0 és 1 Boole értékek)
korlátok= ítéletkalkulusbeli relációk
következtetési mechanizmus= MI SAT-módszerek (SAT — Boole kielégíthetőség)

OR ADAI... ...

ResolutionCSPSimplex ... ...

CLP(H)
CLP(R)

Logic
Programming

Prolog =
CLP(FD)

Mercury = CLP(0)

Példa: CLP(MiniNat)

Egy miniatűr kvázi-CLP nyelv természetes számokra
(Motiváció: a CLP alapelvek és egyben a haladó Prolog lehetőségek bemutatása.)

• Tartomány: Nem negatív egészek

• Függvények:

+ - *

• Korlát-relációk:

= < > =< >=

• Korlát-megoldó algoritmus:

SICStus korutin-kiterjesztésén alapul

A Prologba ágyazás szintaxisa:

{ Korlát} a Korlát felvétele
({X} szintaktikus édesítőszer, ekvivalens a’{}’(X) kifejezéssel.)

Példafutás

| ?- {X+Y = 2}.
X = 2, Y = 0 ? ;
X = 1, Y = 1 ? ;
X = 0, Y = 2 ? ;
no
| ?- {2 * X+3* Y=8}.
X = 4, Y = 0 ? ;
X = 1, Y = 2 ? ;
no
| ?- {X * 2+1=28}.
no
| ?- {X * X+Y* Y=25, X > Y}.
X = 5, Y = 0 ? ;
X = 4, Y = 3 ? ;
no

Prolog háttér: blokkolás, korutinszervezés

Blokk-deklarációk SICStusban
Egy eljárásra előírhatjuk, hogy mindaddig, amíg egy ún. blokkolási feltétel fennáll,
az eljárás függesztődjék fel. Példa:

:- block p(-, ?, -, ?, ?).

Jelentése: ha az első és a harmadik argumentum is behelyettesítetlen változó
(blokkolási feltétel), akkor ap hívás felfüggesztődik.

Ugyanarra az eljárásra több vagylagos feltétel is szerepelhet, pl.
:- block p(-, ?), p(?, -).

Blokk-deklarációk haszna

• Adatfolyam-programozás (lásd Hamming probléma, Prolog jegyzet)

• Generál és ellen̋oriz programok gyorsítása

• Végtelen választási pontok kiküszöbölése

Listák biztonságos összefűzése blokk-deklaráció segítségével

:- block app(-, ?, -).
% blokkol, ha az els ő és a harmadik argumentum
% egyaránt behelyettesítetlen
app([], L, L).
app([X|L1], L2, [X|L3]) :-

app(L1, L2, L3).

| ?- app(L1, L2, L3).
user:app(L1,L2,L3) ? ;
no
| ?- app(L1, L2, L3), L3 = [a|L4].
L1 = [], L2 = [a|L4], L3 = [a|L4] ? ;
L1 = [a|_A], L3 = [a|L4], user:app(_A,L2,L4) ? ;
no

Listák biztonságos összefűzése, nyomkövetés

:- block app(-, ?, -).
% blokkol, ha az els ő és a harmadik argumentum
% egyaránt behelyettesítetlen
app([], L, L).
app([X|L1], L2, [X|L3]) :-

app(L1, L2, L3).

| ?- trace, app(L1, L2, L3), L3 = [a|L4], L4 = [].
% The debugger will first creep -- showing everything (trace )

- - Block: app(_1012,_532,_1018)
1 1 Call: _1018=[a|_622] ?
- - Unblock: app(_1012,_532,[a|_622])
2 2 Call: app(_1012,_532,[a|_622]) ?

? 2 2 Exit: app([],[a|_622],[a|_622]) ?
? 1 1 Exit: [a|_622]=[a|_622] ?

3 1 Call: _622=[] ?
3 1 Exit: []=[] ?

L1 = [], L2 = [a], L3 = [a], L4 = [] ? ;
1 1 Redo: [a|_622]=[a|_622] ?
2 2 Redo: app([],[a|_622],[a|_622]) ?
- - Block: app(_2098,_532,_2104)
2 2 Exit: app([a|_2098],_532,[a|_2104]) ? &

Blocked goals:
1 (_2098): user:app(_2098,_532,_2104)
2 (_2104): user:app(_2098,_532,_2104)

2 2 Exit: app([a|_2098],_532,[a|_2104]) ?
1 1 Exit: [a|_2104]=[a|_2104] ?
4 1 Call: _2104=[] ?
- - Unblock: app(_2098,_532,[])
5 2 Call: app(_2098,_532,[]) ?

? 5 2 Exit: app([],[],[]) ?
? 4 1 Exit: []=[] ?
L1 = [a], L2 = [], L3 = [a], L4 = [] ? ;

4 1 Redo: []=[] ?
5 2 Redo: app([],[],[]) ?
5 2 Fail: app(_2098,_532,[]) ?
4 1 Fail: _2104=[] ?

no

Példa korutinszervezésre: többirányú összeadás

% plusz(X, Y, Z): X+Y=Z, ahol X, Y és Z természetes számok.
% Bármelyik argumentum lehet behelyettesítetlen.
plusz(X, Y, Z) :-

app(A, B, C),
len(A, X),
len(B, Y),
len(C, Z).

% L hossza Len.
len(L, Len) :-

len(L, 0, Len).

:- block len(-, ?, -).
% L lista hossza Len-Len0. Len0 mindig ismert.
len(L, Len0, Len) :-

nonvar(Len), !, Len1 is Len-Len0,
length(L, Len1).

len([_|L], Len0, Len) :-
Len1 is Len0+1, len(L, Len1, Len).

len([], Len, Len).

| ?- plusz(X, Y, 2).
X = 0, Y = 2 ? ;
X = 1, Y = 1 ? ;
X = 2, Y = 0 ? ;
no
| ?- plusz(X, X, 8).
X = 4 ? ;
no
| ?- plusz(X, 1, Y), plusz(X, Y, 20).
no

További korutinszervező eljárások

Hívások késleltetése

freeze(X, Hivas)
Hivas t felfüggeszti mindaddig, amígX behelyettesítetlen változó.

dif(X, Y)
X ésY nem egyesíthető. Mindaddig felfüggesztődik, amíg ez el nem dönthető.

when(Feltétel, Hívás)
Blokkolja aHívás t mindaddig, amíg aFeltétel igazzá nem válik. Itt a
Feltétel egy (nagyon) leegyszerűsített Prolog cél, amelynek szintaxisa:

CONDITION ::= nonvar(X) | ground(X) | ?=(X,Y) |
CONDITION, CONDITION |
CONDITION; CONDITION

(ground(X) jelentése:X, tömör, azaz nem tartalmaz (behelyettesítetlen) változót
?=(X,Y) jelentése:X ésY egyesíthet̋osége eldönthető.)
Példa (process csak akkor hívódik meg, haT tömör, és vagyX nem változó, vagy
X ésY egyesíthet̋osége eldönthető):

| ?- when( ((nonvar(X);?=(X,Y)),ground(T)),
process(X,Y,T)).

A dif eljárás awhen segítségével definiálható:

dif(X, Y) :- when(?=(X,Y), X\==Y).

Késleltetett hívások lekérdezése

frozen(X, Hivas)
Az X változó miatt felfüggesztett hívás(oka)t egyesítiHivas -sal.

call_residue_vars(Hivas, Valtozok)
Hivas -t végrehajtja, és aValtozok listában visszaadja mindazokat az új (a
Hivas alatt létrejött) változókat, amelyekre vonatkoznak felfüggesztett hívások. Pl.

| ?- call_residue_vars((dif(X,f(Y)), X=f(Z)), Vars).

X = f(Z),
Vars = [Z,Y],
prolog:dif(f(Z),f(Y)) ?

Többirányú összeadáswhen segítségével

:- use_module(library(between)).

% app(L1, L2, L3): L1 és L2 összefűzöttje L3.
% ahol L1, L2 és L3 1-es számokból álló listák.
app([], L, L).
app([1|L1], L2, [1|L3]) :-

when((nonvar(L1);nonvar(L3)),
app(L1, L2, L3)).

len(L, Len) :-
when(ground(L), length(L, Len)),
when(nonvar(Len), findall(1, between(1, Len, _), L)).

% X+Y=Z, ahol X, Y és Z természetes számok.
% Bármelyik argumentum lehet behelyettesítetlen.
plusz(X, Y, Z) :-

app(A, B, C),
len(A, X),
len(B, Y),
len(C, Z).

| ?- plusz(X, Y, 2).
X = 0, Y = 2 ? ;
X = 1, Y = 1 ? ;
X = 2, Y = 0 ? ;
no
| ?- plusz(X, X, 8).
X = 4 ? ;
no
| ?- plusz(X, 1, Y), plusz(X, Y, 20).
no

CLP(MiniNat) megvalósítása

Számábrázolás

• A korábbiplusz/3 eljárásban egyN elemű listával ábrázoltuk azN számot
(a listaelemek érdektelenek, behelyettesítetlen változók vagy 1-esek)

• Példa: a2 szám ábrázolása:[_,_] ≡ .(_,.(_,[])) .

• Hagyjuk el a felesleges listaelemeket, akkor a2 szám ábrázolása:.(.([])) .

• Itt a [] jelenti a 0 számot, a.(X) struktúra azX szám rákövetkez̋ojét (a nála
1-gyel nagyobb számot).

• Ez tulajdonképpen a Peano féle számábrázolás, ha a. /1 helyett azs/1
funktort, a[] helyett a0 konstanst használjuk.

• A CLP(MiniNat) megvalósításában a Peano számábrázolást használjuk, tehát;0
= 0; 1 = s(0); 3 = s(s(s(0))) stb.

Összeadás és kivonás

% plusz(X, Y, Z): X+Y=Z (Peano számokkal).
:- block plusz(-, ?, -).
plusz(0, Y, Y).
plusz(s(X), Y, s(Z)) :-

plusz(X, Y, Z).

% +(X, Y, Z): X+Y=Z (Peano számokkal). Hatékonyabb, mert
% továbblép, ha bármelyik argumentum behelyettesített.
:- block +(-, -, -).
+(X, Y, Z) :-

var(X), !, plusz(Y, X, Z). % \+((var(Y),var(Z)))
+(X, Y, Z) :-

/ * nonvar(X), * / plusz(X, Y, Z).

% X-Y=Z (Peano számokkal).
-(X, Y, Z) :-

+(Y, Z, X).

CLP(MiniNat) megvalósítása (folyt.)

A szorzás művelet megvalósítási elvei:

• Felfüggesztjük mindaddig, míg legalább egy tényező vagy a szorzat ismertté
nem válik.

• Ha az egyik tényez̋o ismert, visszavezetjük ismételt összeadásra.

• Ha a szorzat ismert (N ), az egyik tényez̋ore végigpróbáljuk az 1,2,. . .N

értékeket, ezáltal ismételt összeadásra visszavezethetővé tesszük.

% X* Y=Z. Blokkol, ha nincs tömör argumentuma.

* (X, Y, Z) :-
when( (ground(X);ground(Y);ground(Z)),

szorzat(X, Y, Z)).

% X* Y=Z, ahol legalább az egyik argumentum tömör.
szorzat(X, Y, Z) :-

( ground(X) -> szor(X, Y, Z)
; ground(Y) -> szor(Y, X, Z)
; / * Z tömör! * /

Z == 0 -> szorzatuk_nulla(X, Y)
; X = s(_), +(X, _, Z),

% X =< Z, vö. between(1, Z, X)
szor(X, Y, Z)

).

% X* Y=0.
szorzatuk_nulla(X, Y) :-

( X = 0 ; Y = 0 ).

% szor(X, Y, Z): X * Y=Z, X tömör.
% Y-nak az (ismert) X-szeres összeadása adja ki Z-t.
szor(0, _X, 0).
szor(s(X), Y, Z) :-

szor(X, Y, Z1),
+(Z1, Y, Z).

CLP(MiniNat) megvalósítása: (folyt. 2)

A korlátok végrehajtása

• A funkcionális alakban megadott korlátokat a+ /3, - /3, * /3
hívásokból álló célsorozattá alakítjuk, majd ezt a célsorozatot meghívjuk.

• Például a{X * Y+2=Z} korlát lefordított alakja:

* (X, Y, _A),+(_A, s(s(0)), Z) ,

• Az {X =< Y} korlátot az{X+_ = Y} korlátra, az{X < Y} korlátot pedig
az{X+s(_) = Y} korlátra vezetjük vissza

% {Korlat}: Korlat fennáll.
{Korlat} :-

korlat_cel(Korlat, Cel), call(Cel).

Korlátok fordítása

% korlat_cel(Korlat, Cel): Korlat végrehajtható
% alakja a Cel célsorozat.
korlat_cel(Kif1=Kif2, (C1,C2)) :-

kiertekel(Kif1, E, C1), % Kif1 értékét E-ben
% el őállító cél C1

kiertekel(Kif2, E, C2).
korlat_cel(Kif1 =< Kif2, Cel) :-

korlat_cel(Kif1+_ = Kif2, Cel).
korlat_cel(Kif1 < Kif2, Cel) :-

korlat_cel(s(Kif1) =< Kif2, Cel).
korlat_cel(Kif1 >= Kif2, Cel) :-

korlat_cel(Kif2 =< Kif1, Cel).
korlat_cel(Kif1 > Kif2, Cel) :-

korlat_cel(Kif2 < Kif1, Cel).
korlat_cel((K1,K2), (C1,C2)) :-

korlat_cel(K1, C1), korlat_cel(K2, C2).

CLP(MiniNat) megvalósítása: (folyt. 3)

Kifejezések fordítása

• EgyKif1 Op Kif2 kifejezés lefordított alakja egy három részből álló
célsorozat, amely egyE változóban állítja el̋o a kifejezés eredményét:

– els̋o rész:Kif1 értékét pl.A-ban el̋oállító cél(sororzat).

– második rész:Kif2 értékét pl.B-ban el̋oállító cél(sororzat).

– harmadik rész: azOp(A, B, E) hívás (aholOp a+, -, * jelek
egyike).

• Egy szám lefordított formája az̋o Peano alakja.

• Minden egyéb (változó, vagy már Peano alakú szám) változatlan marad a
fordításkor.

% kiertekel(Kif, E, Cel): A Kif aritmetikai kifejezés
% értékét E-ben el őállító cél Cel.
% Kif egészekb ől a +, -, és * operátorokkal épül fel.
kiertekel(Kif, E, (C1,C2,Rel)) :-

nonvar(Kif),
Kif =.. [Op,Kif1,Kif2], !,
kiertekel(Kif1, E1, C1),
kiertekel(Kif2, E2, C2),
Rel =.. [Op,E1,E2,E].

kiertekel(N, Kif, true) :-
number(N), !,
int_to_peano(N, Kif).

kiertekel(Kif, Kif, true).

% int_to_peano(N, P): N természetes szám Peano alakja P.
int_to_peano(0, 0).
int_to_peano(N, s(P)) :-

N > 0, N1 is N-1,
int_to_peano(N1, P).

Prolog háttér: kifejezések testreszabott kiírása

print/1
Alapértelmezésben azonoswrite -tal. Ha a felhasználó definiál egyportray/1
eljárást, akkor a rendszer minden aprint -tel kinyomtatandó részkifejezésre
meghívjaportray -t. Ennek sikere esetén feltételezi, hogy a kiírás megtörtént,
meghiúsulás esetén maga írja ki a részkifejezést.
A rendszer aprint eljárást használja a változó-behelyettesítések és a nyomkövetés
kiírására!

portray/1
Igaz, haKif kifejezést a Prolog rendszernek nem kell kiírnia. Alkalmasformában
kiírja aKif kifejezést.
Ez egy felhasználó által definiálandó (kampó) eljárás (hook predicate).

Példa: mátrixok kiíratása

portray(Matrix) :-
Matrix = [[_|_]|_],
( member(Row, Matrix), nl, print(Row), fail
; true
).

| ?- X = [[1,2,3],[4,5,6]].
X =
[1,2,3]
[4,5,6] ?

Példa testreszabott kiíratásra: Peano számok

% Peano számok kiírásának formázása
user:portray(Peano) :-

peano_to_int(Peano, 0, N), write(N).

% A Peano Peano-szám értéke N-N0.
peano_to_int(Peano, N0, N) :-

nonvar(Peano),
( Peano == 0 -> N = N0
; Peano = s(P),

N1 is N0+1,
peano_to_int(P, N1, N)

).

% felfüggesztett célok kiíratásának formázása
user:portray(user:Rel) :-

Rel =.. [Pred,A,B,C],
predikatum_operator(Pred, Op),
Fun =.. [Op,A,B],
print({Fun=C}).

predikatum_operator(plusz, +).
predikatum_operator(+, +).
predikatum_operator( * , * ).

Prolog háttér: programok előfeldolgozása

Kampó (Hook, callback) eljárások a fordítási idejű átalakításhoz:

• user:term_expansion(+Kif, ..., -Klózok, ...) : (közelít̋o
leírás:) Minden betölt̋o eljárás (consult, compile stb.) által beolvasott
kifejezésre a rendszer meghívja. A kimenő paraméterben várja a transzformált
alakot (lehet lista is). Meghiúsulás esetén változtatás nélkül veszi fel a
kifejezést klózként.

• M:goal_expansion(+Cél, +Layout, +Modul, -ÚjCél,
-ÚjLayout) : Minden a beolvasott programban (vagy feltett kérdésben)
előforduló részcélra meghívja a rendszer. A kimenő paraméterekben várja a
transzformált alakot (lehet konjunkció). Meghiúsulás esetén változtatás nélkül
hagyja a célt. (Ha a forrásszintű nyomkövetés nem fontos,ÚjLayout lehet
[] .)

CLP(MiniNat) továbbfejlesztésegoal_expansion használatával

• A funkcionális alak átalakítása a betöltés alatt is elvégezhet̋o (kompilálás):

goal_expansion({Korlat}, _LO, _Module, Cel, / * ÚjLO* / []) :-
korlat_cel(Korlat, Cel).

• Célszerű a generált célsorozatból atrue hívásokat kihagyni.

% osszetett(C1, C2, C): C a C1 és C2 célok konjunkciója.
osszetett(true, Cel0, Cel) :- !, Cel = Cel0.
osszetett(Cel0, true, Cel) :- !, Cel = Cel0.
osszetett(Cel1, Cel2, (Cel1,Cel2)).

• A fenti eljárást használjuk a konjunkciók helyett, pl:

korlat_cel((K1,K2), C12) :-
korlat_cel(K1, C1), korlat_cel(K2, C2),
osszetett(C1, C2, C12).

Megjegyzés: a faktoriális példában ez a kompilálás 6-7% gyorsulást jelent

Előfeldolgozás a faktoriális példa esetén

• A faktoriális példa betöltött alakja :

fact(0, s(0)).
fact(N, F) :-

+(s(0), _, N), % N >= 1
-(N, s(0), N1), % N1 = N-1

* (N, F1, F), % F = N * F1
fact(N1, F1).

• Vigyázat! Az így el̋oálló kód már nem foglalkozik a számok Peano-alakra
hozásával:

| ?- fact(N, 6). --> no
| ?- {F=6}, fact(N, F). --> F = 6, N = 3 ? ; no

CLP(MiniNat) használata — példák

(Kompilálás nélkül)

:- block fact(-,-).
fact(N, F) :-

{N = 0, F = 1}.
fact(N, F) :-

{N >= 1, N1 = N-1},
fact(N1, F1),
{F = N * F1}.

| ?- fact(6, F).
F = 720 ? ; no

| ?- fact(8, F).
F = 40320 ? ; no

| ?- fact(F, 6).
F = 3 ? ; no

| ?- fact(F, 24).
F = 4 ? ;
! Resource error: insufficient memory

| ?- fact(F, 12).
no

| ?- fact(F, 15).
! Resource error: insufficient memory

| ?- {X * X+Y* Y=25, X>Y}.
X = 4, Y = 3 ? ;
X = 5, Y = 0 ? ;
X = 5, Y = 0 ? ;
no

CLP(MiniNat) javított változatai

A nulla szorzat problémája

| ?- {X * X=0}.
X = 0 ? ; X = 0 ? ; no

A probléma 1. javítása

% X* Y=0, ahol X és Y Peano számok.
szorzatuk_nulla(X, Y) :-

( X = 0
; X \== Y, Y = 0
).

| ?- {X * X=0}.
X = 0 ? ; no

| ?- {X * Y=0}, X=Y.
X = 0, Y = 0 ? ;
X = 0, Y = 0 ? ; no

A probléma 2. javítása

% X* Y=0, ahol X és Y Peano számok.
szorzatuk_nulla(X, Y) :-

( X = 0
; dif(X, 0), Y = 0
).

| ?- {X * Y=0}, X=Y.
X = 0, Y = 0 ? ; no

CLP(MiniNat) javított változatai (folyt)

Az erőforrás probléma

• A fact(N, 11) hívás a második klózzal illesztve a{11=N * F1} feltételre
vezet̋odik vissza. Ez két megoldást generál (N=1,F1=11 , ill. N=11,F1=1 .
Ezekre a behelyettesítésekre felébred a rekurzívfact hívás el̋oször a
fact(0,11) majd afact(10,1) paraméterekkel.

• A fact/2 második klóza ez utóbbit mohón értékeli ki: kiszámolja10! -t, és
csak ezután egyesíti1-gyel. Azonban a10! kiszámolásához (Peano számként)
sok idő és memória kell:-( .

• A probléma javítása: a szorzat-feltételt tegyük a rekurzívfact/2 hívás elé.
Egy további gyorsítási lehetőség aredundánskorlátok alkalmazása.

:- block fact(-,-).
fact(N, F) :- {N = 0, F = 1}.
fact(N, F) :-

{N >= 1, N1 = N-1, F = N * F1},
{F1 >= N1} % redundáns korlát
fact(N1, F1).

| ?- fact(N, 24). --------> N = 4 ? ; no

• Azonban az alábbi cél futása még így is kivárhatatlan . . .

| ?- fact(N, 5040). --------> N = 6 ? ;

Megjegyzések

• Egy korlát-programban minél később célszerű választási pontot csinálni.

• Ideálisan csak az összes korlát felvétele után kezdjük meg akeresést.

• Megoldás: egy külön keresési fázis (az ún. címkézés, labeling):

program :-
korlátok_felvétele(...), labeling([V1, ..., VN]).

• CLP(MiniNat)-ban az ismertetett eszközökkel ez nehezen megoldható, de

• CLP(MiniB) esetén (lásd 1. kis házi feladat) könnyen készíthet̋o ilyen
labeling/1 eljárás.

1. kis házi feladat: CLP(MiniB) megvalósítása

CLP(MiniB) jellemzése

• Tartomány: logikai értékek (1 és0, igaz és hamis)

• Függvények(egyben korlát-relációk):
˜ P P hamis (negáció).
P * Q PésQmindegyike igaz (konjunkció).
P + Q PésQ legalább egyike igaz (diszjunkció).
P # Q PésQpontosan egyike igaz (kizáró vagy).
P =\= Q Ugyanaz mintP # Q.
P =:= Q Ugyanaz mint~(P # Q) .

A megvalósítandó eljárások

• sat( Kif) , aholKif változókból, a0, 1 konstansokból a fenti müveletekkel
felépített logikai kifejezés. Jelentése: AKif logikai kifejezés igaz. Asat/1
eljárás ne hozzon létre választási pontot! A benne szereplő változók
behelyettesítése esetén minél előbb ébredjen fel, és végezze el a megfelelő
következtetéseket (lásd a példákat alább)!

• count( Es, N) , aholEsegy (változó-)lista,N adott természetes szám.
Jelentése: AzEs listában pontosanN olyan elem van, amelynek értéke 1.

• labeling( Változók) . Behelyettesíti aVáltozókat0, 1 értekekre. Visszalépés
esetén felsorolja az összes lehetséges értéket.

Futási példák

| ?- sat(A * B =:= (~A)+B).
---> <...felfüggesztett célok...> ? ; no

| ?- sat(A * B =:= (~A)+B), labeling([A,B]).
---> A = 1, B = 0 ? ; A = 1, B = 1 ? ; no

| ?- sat((A+B) * C=\=A * C+B), sat(A * B).
---> A = 1, B = 1, C = 0 ? ; no

| ?- count([A,A,B], 2). ---> <...felfüggesztett célok...> ? ; no
| ?- count([A,A,B], 2), labeling([A]).

---> A = 1, B = 0 ? ; no
| ?- count([A,A,B,B], 3), labeling([A,B]).

---> no
| ?- sat(~A =:= A). ---> no

1. kis házi feladat: egy kis segítség

:- op(100, fx, ~).

~(A, B) :-
when( (nonvar(A); nonvar(B); ?=(A,B)),

not(A,B)
).

not(A, NA) :-
( nonvar(A) -> NA is 1-A
; nonvar(NA) -> A is 1-NA
; A == NA -> fail
).

| ?- trace, ~(A, A).
1 1 Call: ~(A,A) ?
2 2 Call: when((nonvar(A);nonvar(A);?=(A,A)),not(A,A)) ?
3 3 Call: not(A,A) ?
4 4 Call: nonvar(A) ?
4 4 Fail: nonvar(A) ?
5 4 Call: nonvar(A) ?
5 4 Fail: nonvar(A) ?
6 4 Call: A==A ?
6 4 Exit: A==A ?
3 3 Fail: not(A,A) ?
2 2 Fail: when((nonvar(A);nonvar(A);?=(A,A)),not(A,A)) ?
1 1 Fail: ~(A,A) ?

no
| ?- sat(A * A=:=B).

B = A ? ; no
| ?- sat(A#A=:=B).

B = 0 ? ; no
| ?- sat(A+B=:=C), A=B.

B = A, C = A ? ; no

A SICStus clp(Q,R) könyvtárak

A clpq /clpr könyvtárak

• Tartomány:

– clpr : lebeg̋opontos számok

– clpq : racionális számok

• Függvények:

+ - * / min max pow exp (kétargumentumúak,pow ≡ exp ),

+ - abs sin cos tan (egyargumentumúak).

• Korlát-relációk:

= =:= < > =< >= =\= (= ≡ =:= )

• Primitív korlátok (korlát tár elemei):

lineáris kifejezéseket tartalmazó relációk

• Korlát-megoldó algoritmus:

lineáris programozási módszerek: Gauss elimináció, szimplex módszer

A könyvtár betöltése:

• use_module(library(clpq)) , vagy

• use_module(library(clpr))

A fő beépített eljárás

• { Korlát } , aholKorlát változókból és (egész vagy lebegőpontos) számokból
a fenti műveletekkel felépített reláció, vagy ilyen relációknak a vessz̋o (, )
operátorral képzett konjunkciója.

Példafutás a SICStusclpq könyvtárával

Példafutás

| ?- use_module(library(clpq)).
{loading .../library/clpq.ql...}
...

| ?- {X=Y+4, Y=Z-1, Z=2 * X-9}.
X = 6, Y = 2, Z = 3 ? % lineáris egyenlet

| ?- {X+Y+9<4 * Z, 2 * X=Y+2, 2 * X+4* Z=36}.
% lineáris egyenl őtlenség

{X<29/5}, {Y= -2+2 * X}, {Z=9-1/2 * X} ?
% az eredmény: a tár állapota

| ?- {(Y+X) * (X+Y)/X = Y * Y/X+100}.
{X=100-2 * Y} ? % lineárissá egyszerűsíthet ő

| ?- {(Y+X) * (X+Y) = Y * Y+100* X}.
% így már nem lineáris

clpq:{2 * (X * Y)-100 * X+X^2=0} ?
% a clpq modul-prefix jelzi,
% hogy felfüggesztett összetett
% hívásról van szó

| ?- {exp(X+Y+1,2) = 3 * X* X+Y* Y}.
% nem lineáris...

clpq:{1+2 * X+2* (Y * X)-2 * X^2+2 * Y=0} ?

| ?- {exp(X+Y+1,2) = 3 * X* X+Y* Y}, X=Y.
X = -1/4, Y = -1/4 ? % így már igen...

| ?- {2 = exp(8, X)}. % nem-lineárisak is
% megoldhatók

X = 1/3 ?

Összetett korlátok kezelése CLP(Q)-ban

Példa várakozó ágensre

| ?- {X =< Y}, {X * (Y+1) > X * X+Z},
( Z = X * (Y-X), {Y < 0}
; Y = X
).

Y = X, {X-Z>0} ? ; no

A végrehajtás lépései

| ?- {X =< Y}, {X * (Y+1) > X * X+Z}.
{X-Y=<0}, clpq:{Z-X-Y * X+X^2<0} ?

| ?- {X =< Y}, {X * (Y+1) > X * X+Z}, Z = X * (Y-X).
Z = X* (Y-X), {X-Y=<0}, {X>0} ?

| ?- {X =< Y}, {X * (Y+1) > X * X+Z}, Z = X * (Y-X), {Y < 0}.
no

| ?- {X =< Y}, {X * (Y+1) > X * X+Z}, Y = X.
Y = X, {X-Z>0} ?

Példa egylehetségeserősítési lépésre

• A tár tartalma:X > 3.

• A végrehajtandó összetett korlát:Y > X* X.

• A korlátot a CLP megoldó nem tudja felvenni a tárba, de egykövetkezményét,
pl. azY > 9 korlátot felvehetné!

• Az erősítés után az eredeti összetett korlát továbbra is démonként kell lebegjen!

• Fontos megjegyzés:a CLP(Q/R) rendszernem hajtja végre a fenti
következtetést, és általánosan semmiféle erősítést nem végez.

Egy összetettebb példa: hiteltörlesztés

% Hiteltörlesztés számítása: P összegű hitelt
% Time hónapon át évi IntRate kamat mellett havi MP
% részletekben törlesztve Bal a maradványösszeg.
mortgage(P, Time, IntRate, Bal, MP):-

{Time > 0, Time =< 1,
Bal = P * (1+Time * IntRate/1200)-Time * MP}.

mortgage(P, Time, IntRate, Bal, MP):-
{Time > 1},
mortgage(P * (1+IntRate/1200)-MP,

Time-1, IntRate, Bal, MP).

| ?- mortgage(100000,180,12,0,MP).
% 100000 Ft hitelt 180
% hónap alatt törleszt 12%-os
% kamatra, mi a havi részlet?

MP = 1200.1681 ?

| ?- mortgage(P,180,12,0,1200).
% ugyanez visszafelé

P = 99985.9968 ?

| ?- mortgage(100000,Time,12,0,1300).
% 1300 Ft a törleszt őrészlet,
% mi a törlesztési id ő?

Time = 147.3645 ?

| ?- mortgage(P,180,12,Bal,MP).

{MP=0.0120 * P-0.0020 * Bal} ?

| ?- mortgage(P,180,12,Bal,MP), ordering([P,Bal,MP]).

{P=0.1668 * Bal+83.3217 * MP} ?

További könyvtári eljárások

entailed(Korlát) — Korlát levezethet̋o a jelenlegi tárból.

inf(Kif, Inf) ill. sup(Kif, Sup) — kiszámoljaKif infímumát ill.
szuprémumát, és egyesítiInf -fel ill. Sup-pal. Példa:

| ?- { 2 * X+Y =< 16, X+2 * Y =< 11, X+3 * Y =< 15,
Z = 30* X+50* Y

}, sup(Z, Sup).

Sup = 310, {....}

minimize(Kif) ill. maximize(Kif) — kiszámoljaKif infimumát ill.
szuprémumát, és egyenlővé tesziKif -fel. Példa:

| ?- { 2 * X+Y =< 16, X+2 * Y =< 11, X+3 * Y =< 15,
Z = 30* X+50* Y

}, maximize(Z).

X = 7, Y = 2, Z = 310

bb_inf(Egészek, Kif, Inf) — kiszámoljaKif infimumát, azzal a további
feltétellel, hogy az Egészek listában levő minden változó egész (ún. „Mixed Integer
Optimisation Problem”).

| ?- {X >= 0.5, Y >= 0.5}, inf(X+Y, I).

I = 1, {Y>=1/2}, {X>=1/2} ?

| ?- {X >= 0.5, Y >= 0.5}, bb_inf([X,Y], X+Y, I).

I = 2, {X>=1/2}, {Y>=1/2} ?

ordering(V1 < V2) — A V1 változó el̋obb szerepeljen az eredmény-korlátban
mint aV2 változó.
ordering([V1,V2,...]) — V1, ... ebben a sorrendben szerepeljen az
eredmény-korlátban.

További eljárások (lásd kézikönyv):bb_inf/5 , dump/3 ,
projecting_assert/1 ,

Széls̋oérték-számítás grafikus illusztrálása

2x+y=<16

x+3y=<15x+2y
=<11

310=30x+50y

| ?- { 2 * X+Y =< 16, X+2 * Y =< 11, X+3 * Y =< 15,
Z = 30* X+50* Y

}, sup(Z, Sup).

Sup = 310, {Z=30 * X+50* Y},
{X+1/2 * Y=<8}, {X+3 * Y=<15}, {X+2 * Y=<11}

További részletek

Projekció

% Az (X,Y) pont az (1,2) (1,4) (2,4) pontok
% által kifeszített háromszögben van.
hszogben(X,Y) :-

{ X=1 * L1+1 * L2+2 * L3,
Y=2* L1+4 * L2+4 * L3,
L1+L2+L3=1, L1>=0, L2>=0, L3>=0 }.

| ?- hszogben(X,Y).
{Y=<4}, {X>=1}, {X-1/2 * Y=<0} ?

| ?- hszogben(_, Y).
{Y=<4}, {Y>=2} ?

| ?- hszogben(X, _).
{X>=1}, {X=<2} ?

Bels̋o ábrázolás
clpr — lebeg̋opontos szám;clpq —- rat( Számláló, Nevező) , aholSzámláló
ésNevezőrelatív prímek. Példáulclpq -ban:

| ?- {X=0.5}, X=0.5.
no
| ?- {X=0.5}, X=1/2.
no
| ?- {X=0.5}, X=rat(2,4).
no
| ?- {X=0.5}, X=rat(1,2).
X = 1/2 ?
| ?- {X=5}, X=5.
no
| ?- {X=5}, X=rat(5,1).
X = 5 ?

Egy nagyobb CLP(Q) feladat: Tökéletes téglalapok

A feladat

• egy olyan téglalap keresése

• amely kirakható páronként különböző oldalú négyzetekb̋ol

Egy megoldás
(a legkevesebb, 9 darab négyzet felhasználásával)

18
15

14

4
78

9
10

32

33

1

Tökéletes téglalapok — CLP(Q) megoldás

% Colmerauer A.: An Introduction to Prolog III,
% Communications of the ACM, 33(7), 69-90, 1990.

% Rectangle 1 x Width is covered by distinct
% squares with sizes Ss.
filled_rectangle(Width, Ss) :-

{ Width >= 1 }, distinct_squares(Ss),
filled_hole([-1,Width,1], _, Ss, []).

% distinct_squares(Ss): All elements of Ss are distinct.
distinct_squares([]).
distinct_squares([S|Ss]) :-

{ S > 0 }, outof(Ss, S), distinct_squares(Ss).

outof([], _).
outof([S|Ss], S0) :- { S =\= S0 }, outof(Ss, S0).

% filled_hole(L0, L, Ss0, Ss): Hole in line L0
% filled with squares Ss0-Ss (diff list) gives line L.
% Def: h(L): sum of lengths of vertical segments in L.
% Pre: All elements of L0 except the first >= 0.
% Post: All elems in L >=0, h(L0) = h(L).
filled_hole(L, L, Ss, Ss) :-

L = [V|_], {V >= 0}.
filled_hole([V|HL], L, [S|Ss0], Ss) :-

{ V < 0 }, placed_square(S, HL, L1),
filled_hole(L1, L2, Ss0, Ss1), { V1=V+S },
filled_hole([V1,S|L2], L, Ss1, Ss).

% placed_square(S, HL, L): placing a square size S on
% horizontal line HL gives (vertical) line L.
% Pre: all elems in HL >=0
% Post: all in L except first >=0, h(L) = h(HL)-S.
placed_square(S, [H,V,H1|L], L1) :-

{ S > H, V=0, H2=H+H1 },
placed_square(S, [H2|L], L1).

placed_square(S, [S,V|L], [X|L]) :- { X=V-S }.
placed_square(S, [H|L], [X,Y|L]) :-

{ S < H, X= -S, Y=H-S }.

Tökéletes téglalapok: példafutás

% 600 MHz Pentium III
| ?- length(Ss, N), N > 1, statistics(runtime, _),

filled_rectangle(Width, Ss),
statistics(runtime, [_,MSec]).

N = 9, MSec = 8010, Width = 33/32,
Ss = [15/32,9/16,1/4,7/32,1/8,7/16,1/32,5/16,9/32] ? ;

N = 9, MSec = 1010, Width = 69/61,
Ss = [33/61,36/61,28/61,5/61,2/61,9/61,25/61,7/61,16/ 61] ? ;

N = 9, MSec = 10930, Width = 33/32,
Ss = [9/16,15/32,7/32,1/4,7/16,1/8,5/16,1/32,9/32] ?

Az outof hívás kihagyásával végzett futtatás

Kommentként közöljük az adott ágon generált korlátokat, a redundánsak
elhagyásával.

| ?- filled_rectangle(W, [S1,S2,S3], [eqsq]).

S1 = 1/2, S2 = 1, S3 = 1/2, W = 3/2 ? ; % 3 3 2 2 2 2
% 3 3 2 2 2 2

% {W=S1+S2}, {S2=<1}, {S1=S3}, % 1 1 2 2 2 2
% {S2>=S1+S3}, {S1+S3>=1}. % 1 1 2 2 2 2

S1 = 1, S2 = 1/2, S3 = 1/2, W = 3/2 ? ; % 1 1 1 1 3 3
% 1 1 1 1 3 3

% {W=S1+S2}, {S2=S3}, {S2+S3=<1}, % 1 1 1 1 2 2
% {S2+S3>=S1}, {S1>=1}. % 1 1 1 1 2 2

S1 = 1, S2 = 1, S3 = 1, W = 3 ? ; no

% {W=S1+S2+S3}, {S3=<1}, {S3>=S2}, % 1 1 2 2 3 3
% {S2>=S1}, {S1>=1}. % 1 1 2 2 3 3

Tökéletes téglalapok: választási pontok

Függőleges

V1=<V2 V1>V2

Függ.vál.

V1
V2

Vízszintes

S

H

H1

V

Vízsz. vál.
V=0,
S>H

S=H

S<H

Tökéletes téglalapok: a keresési tér szerkezete

?

?

?

?>0

zsákutca

S4>=S3

S2>=S1

?=0

S2<S1

S4<S3

?

?

Függ. vál.

Vízsz. vál.

Függ. vál.

S1

S1
S2

S1
S2

S3
S4

S2

S6

S5
S4S3

S9 S8

S7

S1

W

1

A CLP(X ) séma

Egy adott CLP(X ) meghatározásakor meg kell adni

• a korlát-következtetés tartományát,

• a korlátok szintaxisát és jelentését (függvények, relációk),

• a korlát-megoldó algoritmust.

A korlátok osztályozása

• egyszerű korlátok— a korlát-megoldó azonnal tudja kezelniőket;

• összetett korlátok— felfüggesztve, démonként várnak arra, hogy a
korlát-megoldónak segíthessenek.

A CLP(X ) korlát-megoldók közös vonása: akorlát tár

• A korlát tárkonzisztenskorlátok halmaza (konjunkciója).

• A korlát tár elemei egyszerű korlátok.

• A közönséges Prolog végrehajtás során a kurrens célsorozatmellett a CLP(X )
rendszer nyilvántartja a korlát tár állapotát:

– amikor a végrehajtás egy egyszerű korláthoz ér, akkor azt amegoldó
megpróbálja hozzávenni a tárhoz;

– ha az új korlát hozzávételével a tár konzisztens marad, akkor ez a redukciós
lépés sikeres és a tár kibővül az új korláttal;

– ha az új korlát hozzávételével a tár inkonzisztenssé válna,akkor (nem kerül
be a tárba és) meghiúsulást, azaz visszalépést okoz;

– visszalépés esetén a korlát tár is visszaáll a korábbi állapotába.

• a összetett korlátok démonként (ágensként) várakoznak arra, hogy:

a. egyszerű korláttá váljanak

b. a tárat egy egyszerű következményükkel bővíthessék (az ún. erősítés)

A korlát logikai programozás elmélete

Egy CLP rendszer

• 〈D,F ,R,S〉

• D: egy tartomány (domain), pl. egészek (N), valósak (R), racionálisak(Q),
Boole értékek (B), listák, füzérek (stringek) (+ a Prolog-fastruktúrák
(Herbrand — H) tartománya)

• F : D-ben definiált függvényjeleknek egy halmaza, pl.+, −, ∗, ∨, ∧

• R: D-ben definiált relációjeleknek (korlátoknak) egy halmaza pl. =, 6=, <, ∈

• S: egy korlát-megoldó algoritmus〈D,F ,R〉-re, azaz aD tartományban az
F ∪R halmazbeli jelekb̋ol felépített korlátokra

CLP szintaxis és deklaratív szemantika

program

• klózok halmaza.

klóz

• szintaxis:P :- G 1, . . . ,Gn, ahol mindegyikGi vagy eljáráshívás, vagy korlát.

• deklaratív olvasat:P igaz, haG1, . . . ,Gn mind igaz.

kérdés

• szintaxis:?- G 1, . . . ,Gn

• válasz egyQkérdésre: korlátoknak egy olyan konjunkciója, amelyből a kérdés
következik.

CLP procedurális szemantika

Végrehajtási állapot

• 〈G, s〉

• G— cél/korlát sorozat

• s — korlát-tár: az eddig felhalmozott egyszerű korlátok konjunkciója
(kezdetben üres)

Szükséges megkülönböztetés

• egyszerű korlát (c ): amit a korlát-tár közvetlenül befogad (F ∪R-től függ)

• összetett korlát (C): a tár nem tudja befogadni, de hathat a tárra

Klózok procedurális olvasata

• P :- G 1, . . . ,Gn jelentése:P megoldásához megoldandóG1, . . . ,Gn.

Végrehajtási invariánsok

• s konzisztens

• G∧ s → Q(Qa kezd̋o kérdés)

Végrehajtás vége

• 〈Ge, se〉, aholGe-re nem alkalmazható egyetlen következtetési lépés sem.

A végrehajtás eredménye

• Az se korlát-tár, vagy annak a kérdésben szereplő változókra való „vetítése” (a
többi változó egzisztenciális kvantálásával).

• A Ge fennmaradó (összetett) korlátok.

A CLP következtetés folyamata

Következtetési lépések

• rezolúció:
〈P & G, s〉 ⇒ 〈G1 & . . . & Gn & G, P = P′ ∧ s〉,

feltéve, hogy a programban van egyP′ :- G 1, . . . ,Gn klóz

• korlát-megoldás:
〈c & G, s〉 ⇒ 〈G, s ∧ c〉

• korlát-er̋osítés:
〈C& G, s〉 ⇒ 〈C′ & G, s ∧ c〉

has-ből következik, hogyCekvivalens (C′ ∧ c )-vel. (C′ = C is lehet.)

Ha a tár inkonzisztensé válna, visszalépés történik.

Példa er̋osítésre

• 〈X > Y* Y & . . . , Y > 3〉 ⇒ 〈X > Y* Y & . . . , Y > 3 ∧ X > 9〉

hiszenX > Y* Y ∧ Y > 3 ⇒ X > 9

• clp(R)-ben nincs ilyen, de clp(FD)-ben van!

Követelmények a korlát megoldó algoritmussal szemben

• teljesség (egyszerű korlátok konjunkciójáról mindig döntse el, hogy
konzisztens-e),

• inkrementalitás (azs tár konzisztenciáját ne bizonyítsa újra),

• a visszalépés támogatása,

• hatékonyság.

38



A clpb könyvtár

• Tartomány: logikai értékek (1 és0, igaz és hamis)

• Függvények(egyben korlát-relációk):
˜ P P hamis (negáció).
P * Q PésQmindegyike igaz (konjunkció).
P + Q PésQ legalább egyike igaz (diszjunkció).
P # Q PésQpontosan egyike igaz (kizáró vagy).
X ˆ P Létezik olyanX, hogyP igaz

(azazP[X/0]+P[X/1] igaz).
P =\= Q Ugyanaz mintP # Q.
P =:= Q Ugyanaz mint~(P # Q) .
P =< Q Ugyanaz mint~P + Q.
P >= Q Ugyanaz mintP + ~Q.
P < Q Ugyanaz mint~P * Q.
P > Q Ugyanaz mintP * ~Q.
card(Is, Es) Az Es listában szereplő igaz értékű kifejezések

száma eleme azIs által jelölt halmaznak (Is
egészek ésTol-Ig szakaszok listája).

• Egyszerű korlátok (korlát tár elemei): tetsz̋oleges korlát (Boole-egyesítők
formájában).

• Korlát-megoldó algoritmus: Boole-egyesítés.

A library(clpb) könyvtár eljárásai

• sat( Kifejezés) , aholKifejezésváltozókból, a0, 1 konstansokból és
atomokból (ún. szimbolikus konstansok) a fenti müveletekkel felépített logikai
kifejezés. HozzávesziKifejezést a korlát-tárhoz.

• taut( Kif, Ért) . Megvizsgálja, hogyKif levezethet̋o-ea tárból, ekkorÉrt=1;
vagy negáltja levezethető-e, ekkorÉrt=0. Egyébként meghiúsul.

• labeling( Változók) . Behelyettesíti aVáltozókat0, 1 értekekre (úgy, hogy a
tár teljesüljön). Visszalépéskor felsorolja az összes lehetséges értéket.

Egyszerű példák

| ?- sat(X + Y). sat(X=\=_A * Y#Y) ?

| ?- sat(x + Y). sat(Y=\=_A * x#x) ?

| ?- taut(_A ^ (X=\=_A * Y#Y) =:= X+Y, T).
T = 1 ?

| ?- sat(A # B =:= 0). B = A ?

| ?- sat(A # B =:= C), A = B. B = A, C = 0 ?

| ?- taut(A =< C, T). no

| ?- sat(A =< B), sat(B =< C), taut(A =< C, T).
T = 1,
sat(A=:=_A * _B* C),
sat(B=:=_B * C) ?

Megjegyzések

• A tár megjelenítése:sat(V =:= Kif) ill. sat(V = \= Kif) aholKif
egy „polinom”, azaz konjunkciókból kizáró vagy (#) művelettel képzett
kifejezés.

• Az atommal jelölt szimbolikus konstansok nem behelyettesíthet̋oek, (legkívül)
univerzálisan kvantifikált változóknak tekinthetők.

| ?- sat(~x+ ~y=:= ~(x * y)). % ∀xy (¬x ∨ ¬y = ¬(x ∧ y))

yes
| ?- sat(~X+ ~Y=:= ~(X * Y)). % ∃?XY(¬X∨ ¬Y = ¬(X∧ Y))

true ? ; no
| ?- sat(x=<y). % ∀xy (x → y)

no
| ?- sat(X=<y). % ∀y∃?X(X → y)

sat(X=:=_A * y) ? ; no

Példa: 1-bites összeadó

| ?- [user].
| adder(X, Y, Sum, Cin, Cout) :-

sat(Sum =:= card([1,3],[X,Y,Cin])),
sat(Cout =:= card([2-3],[X,Y,Cin])).

| {user consulted, 40 msec 576 bytes}

yes
| ?- adder(x, y, Sum, cin, Cout).

sat(Sum=:=cin#x#y),
sat(Cout=:=x * cin#x * y#y * cin) ?

yes
| ?- adder(x, y, Sum, 0, Cout).

sat(Sum=:=x#y),
sat(Cout=:=x * y) ?

yes
| ?- adder(X, Y, 0, Cin, 1), labeling([X,Y,Cin]).

Cin = 0, X = 1, Y = 1 ? ;

Cin = 1, X = 0, Y = 1 ? ;

Cin = 1, X = 1, Y = 0 ? ;

no

Boole-egyesítés

A feladat:

• Adott g ésh logikai kifejezések.

• Keressük ag = h egyenletet megoldó legáltalánosabb egyesítőt (mgu).

• Példa: mgu(X+Y, 1) lehetX = W * Y # Y # 1 (új változó, pl.W, bejöhet).

• Egyszerűsítés: Ag = h egyenlet helyettesíthető azf = 0 egyenlettel, aholf
= g # h .

• Az egyesítés során minden lépésben egyf = 0 formulabeli változót
szeretnénk kifejezni.

Az X változó kifejezése

• LegyenfX(1) azf -ből azX=1, fX(0) azX=0 behelyettesítéssel kapott kifejezés.

• f = 0 kielégíthet̋oségének szükséges feltételefX(1) * fX(0) = 0
kielégíthet̋osége.

• Fejezzük kiX-etfX(0)-val ésfX(1)-gyel úgy, hogyf = 0 legyen!

fX(0) fX(1) X
0 0 bármi (W)
0 1 0
1 0 1
1 1 érdektelen

KeressükX-etX = A* ˜W # B* Walakban!

• Határozzuk megA-t ésB-t fX(0) ésfX(1) függvényeként!

fX(0) fX(1) X A B
0 0 W 0 1
0 1 0 0 0
1 0 1 1 1

Az A = fX(0) ésB = ˜ fX(1) megfeleltetés tűnik a legegyszerűbbnek.

Boole-egyesítés (folyt.)

Az egyesítési algoritmus azf = 0 egyenl̋oségre

• Haf -ben nincs változó, akkor azonosnak kell lennie0-val (különben nem
egyesíthet̋o).

• Helyettesítsünk:X = ˜W* fX(0) # W* ˜ fX(1) (Boole-egyesít̋o)

• Folytassuk az egyesítést azfX(1) * fX(0) = 0 egyenl̋oségre.

Példák

• mgu(X+Y, 0) −→ X = 0, Y = 0 ;

• mgu(X+Y, 1) = mgu(̃ (X+Y) , 0) −→ X = W * Y # Y # 1;

• mgu(X* Y, ˜(X * Z) ) = mgu((X * Y)#(X * Z)#1 , 0) −→ X = 1, Y = ˜Z .

Bels̋o ábrázolás: BDD (Boolean/Binary Decision Diagrams)

(Szaggatott vonal: 0 érték, folytonos vonal: 1 érték)

(X+Y) # 1 X * Y # X* Z # 1

Y

1

0

X

1 0

Z

Y

X

Z

Példa: Hibakeresés áramkörben

X

Sum

Cout

Cin

Y

U1

U3
U2

fault([F1,F2,F3,F4,F5], [X,Y,Cin], [Sum,Cout]) :-
sat(

card([0-1],[F1,F2,F3,F4,F5]) *
(F1 + (U1 =:= X * Cin)) *
(F2 + (U2 =:= Y * U3)) *
(F3 + (Cout =:= U1 + U2)) *
(F4 + (U3 =:= X # Cin)) *
(F5 + (Sum =:= Y # U3))

).

| ?- fault(L, [1,1,0], [1,0]).
L = [0,0,0,1,0] ? ; no

| ?- fault(L, [1,0,1], [0,0]).
L = [_A,0,_B,0,0],
sat(_A=\=_B) ? ; no

| ?- fault(L, [1,0,1], [0,0]), labeling(L).
L = [1,0,0,0,0] ? ;
L = [0,0,1,0,0] ? ; no

| ?- fault([0,0,0,0,0], [x,y,cin], [Sum,Cout]).
sat(Cout=:=x * cin#x * y#y * cin),
sat(Sum=:=cin#x#y) ? ; no

Példa: Tranzisztoros áramkör verifikálása

A

B

Out
X

1

0

n(D, G, S) :- % Gate => Drain = Source
sat( G * D =:= G* S).

p(D, G, S) :- % ~ Gate => Drain = Source
sat( ~G * D =:= ~G* S).

xor(A, B, Out) :-
p(1, A, X),
n(0, A, X),
p(B, A, Out),
n(B, X, Out),
p(A, B, Out),
n(X, B, Out).

| ?- n(D, 1, S). S = D ?

| ?- n(D, 0, S). true ?

| ?- p(D, 0, S). S = D ?

| ?- p(D, 1, S). true ?

| ?- xor(a, b, X). sat(X=:=a#b) ?

Minesweeperclpb -ben

:- use_module([library(clpb),library(lists)]).

mine(Rows, Cols, Mines, Bd) :-
length(Bd, Rows), all_length(Bd, Cols),
append_lists(Bd, All),
sat(card([Mines], All)), play_mine(Bd, []).

all_length([], _).
all_length([L|Ls], Len) :-

length(L, Len), all_length(Ls, Len).

append_lists([], []).
append_lists([L|Ls], Es) :-

append_lists(Ls, Es0), append(L, Es0, Es).

play_mine(Bd, Asked) :-
select_field(Bd, Asked, R, C, E), !,
format(’Row ~w, col ~w (m for mine)? ’, [R,C]),
read(Ans), process_ans(Ans, E, R, C, Bd),
play_mine(Bd, [R-C|Asked]).

play_mine(_Bd, _Asked).

select_field(Bd, Asked, R, C, E) :-
nth(R, Bd, L), nth(C, L, E),
non_member(R-C, Asked), taut(E, 0), !.

select_field(Bd, _Asked, R, C, E) :-
nth(R, Bd, L), nth(C, L, E),
non_member(R-C, Asked), \+ taut(E,1), !.

process_ans(m, 1, _, _, _) :-
format(’Mine!~n’, []), !, fail.

process_ans(Ans, 0, R, C, Bd) :-
integer(Ans), neighbs(n(R, C, Bd), Ns),
sat(card([Ans], Ns)).

neighbs(RCB, N7) :-
neighbour(-1,-1, RCB, [], N0),
neighbour(-1, 0, RCB, N0, N1),
neighbour(-1, 1, RCB, N1, N2),
neighbour( 0,-1, RCB, N2, N3),
neighbour( 0, 1, RCB, N3, N4),
neighbour( 1,-1, RCB, N4, N5),
neighbour( 1, 0, RCB, N5, N6),
neighbour( 1, 1, RCB, N6, N7).

neighbour(ROf, COf, n(R0, C0, Bd), Nbs, [E|Nbs]) :-
R is R0+ROf, C is C0+COf,
nth(R, Bd, Row), nth(C, Row, E), !.

neighbour(_, _, _, Nbs, Nbs).

A SICStusclpfd könyvtár

Tartomány

Egészek (negatívak is) véges (esetleg végtelen) halmaza

Korlátok

• aritmetikai

• halmaz (halmazba tartozás)

• tükrözött

• logikai

• kombinatorikai

• felhasználó által definiált

Egyszerű korlátok

csak a halmaz-korlátok:X ∈ Halmaz

Korlát-megoldó algoritmus

• egyszerű korlátok kezelése triviális;

• a lényeg az összetett korlátokerősítő tevékenysége, ez a Mesterséges
Intelligencia CSP (Constraint Satisfaction Problems) ágának módszerein alapul.

Mir ől lesz szó?

• CSP, mint háttér

• Alapvet̋o (aritmetikai és halmaz-) korlátok

• Tükrözött és logikai korlátok

• Címkéz̋o eljárások

• Kombinatorikai korlátok

• Felhasználó által definiált korlátok: indexikálisok és globális korlátok

• Az FDBG nyomkövet̋o csomag

• Esettanulmányok: négyzetdarabolás, torpedó-, ill, dominó-feladvány

Háttér: CSP (Constraint Satisfaction Problems)

Példafeladat

Az alábbi térkép kiszínezése kék, piros és
sárga színekkel úgy, hogy a szomszédos
országok különböz̋o színűek legyenek, és
ha két ország határán a< jel van, akkor a
két szín ábécé-rendben a megadott módon
kövesse egymást.

Egy lehetséges megoldási folyamat (zárójelben a CSP elnevezések)

1. Minden mez̋oben elhelyezzük a három
lehetséges színt (változók és tartományaik
felvétele). ��

2. Az „A” mező nem lehet kék, mert
annál „B” nem lehetne kisebb. A
„B” nem lehet sárga, mert annál „A”
nem lehetne nagyobb. Az „E” és „D”
mez̋ok hasonlóan szűkíthetők (szűkítés, él-
konzisztencia biztosítása).

3. Ha az „A” mez̋o piros lenne,
akkor mind „B”, mind „D” kék lenne,
ami ellentmondás (globális korlát, ill.
borotválási technika). Tehát „A” sárga.
Emiatt a vele szomszédos „C” és „E” nem
lehet sárga (él-konszitens szűkítés).

���
���
���

���
���
���
���
���
���

���
���
��� ��

��
��

��
��
��

��
��
��
��

���
���
���
���
���
���
���
���

��
��
��

��
��
��

���
���
���

���
���
���

C

D E

<
A

<

B

4. „C” és „D” nem lehet piros, tehát
kék, így „B” csak piros lehet (él-konszitens
szűkítés). Tehát az egyetlen megoldás:
A = sárga, B = piros, C = kék, D = kék,
E = piros.

A CSP problémakör rövid áttekintése

A CSP fogalma

• CSP =(X, D, C)

– X = 〈x1, . . . , xn〉 — változók

– D = 〈D1, . . . , Dn〉 — tartományok, azaz nem üres halmazok

– xi változó aDi véges halmazból (xi tartománya) vehet fel értéket

– C a problémában szereplő korlátok (atomi relációk) halmaza,
argumentumaikX változói (példáulC ∋ c = r(x1, x3), r ⊆ D1 × D3)

• A CSP feladat megoldása: mindenxi változóhoz egyvi ∈ Di értéket kell
rendelni úgy, hogy mindenc ∈ C korlátot egyidejűleg kielégítsünk.

• Definíció: egyc korlát egyxi változójánakdi értékefelesleges, ha nincs ac
többi változójának olyan értékrendszere, amelydi-vel együtt kielégític-t.

• Állítás: felesleges érték elhagyásával (szűkítés) ekvivalens CSP-t kapunk.

• Definíció: egy korlátél-konzisztens(arc consistent), ha egyik változójának
tartományában sincs felesleges érték. A CSPél-konzisztens, ha minden korlátja
él-konzisztens. Az él-konzisztencia szűkítéssel biztosítható.

• Ha minden reláció bináris, a CSP probléma gráffal ábrázolható (változó⇒
csomópont, reláció⇒ él). Az él-konzisztencia elnevezés ebből fakad.

A CSP megoldás folyamata

• felvesszük a változók tartományait;

• felvesszük a korlátokat mint démonokat, amelyek szűkítéssel él-konzisztenciát
biztosítanak;

• többértelműség esetén címkézést (labeling) végzünk:

– kiválasztunk egy változót (pl. a legkisebb tartományút),

– a tartományt két vagy több részre osztjuk (választási pont),

– az egyes választásokat visszalépéses kereséssel bejárjuk(egy tartomány
üresre szűkülése váltja ki a visszalépést).

A térképszínezés mint CSP feladat

Modellezés (leképezés CSP-re)

• változók meghatározása: országonként egy változó, amely az ország színét
jelenti;

• változóértékek kódolása: kék→ 1, piros→ 2, sárga→ 3 (sok CSP
megvalósítás kiköti, hogy a tartományok elemei pl. nem-negatív egészek);

• korlátok meghatározása:

– az el̋oírt < relációk teljesülnek,

– a többi szomszédos ország-pár különböző színű.

A kiinduló korlát-gráf

B

D

C

E

A

{1,2,3} {1,2,3}

{1,2,3}

{1,2,3}

{1,2,3}

A korlát-gráf él-konzisztens szűkítése

B

D

C

E

A

{1,2} {2,3}

{1,2,3}

{2,3}

{1,2}

CLP(FD) = a CSP beágyazása a CLP(X ) sémába

A CSP→ CLP(FD) megfeleltetés

• CSP változó→ CLP változó

• CSP:x tartományaT → CLP: „X in T ” egyszerű korlát.

• CSP korlát→ CLP korlát,általában összetett!

A CLP(FD) korlát-tár

• Tartalma:X in Tartományalakú egyszerű korlátok.

• Tekinthet̋o úgy mint egy hozzárendelés a változók és tartományaik (lehetséges
értékek) között.

• Egyszerű korlát hozzávétele a tárhoz: egy már bennlévő változó tartományának
szűkítése vagy egy új változó-hozzárendelés felvétele.

Összetett CLP(FD) korlátok

• A korlátok többsége démon lesz, hatását akorlát-erősítésen keresztül fejti ki
(〈C, s〉 −→ 〈C ′, s ∧ c〉 ahols |= C ≡ C ′ ∧ c).

• Az erősítés egy egyszerű korlát hozzávételét, azaz a CLP(FD) esetén a tár
szűkítését jelenti.

• A démonok ciklikusan működnek: szűkítenek, elalszanak,aktiválódnak,
szűkítenek, . . . .

• A démonokat a korlátbeli változók tartományának változásaaktiválja.

• Különböz̋o korlátok különböz̋o mértékű szűkítést alkalmazhatnak (a maximális
szűkítés túl drága lehet).

A clpfd könyvtár — alapvető-korlátok

Alapvető aritmetikai korlátok

• Függvények

+ - * / mod min max (kétargumentumúak),

abs (egyargumentumú).

• Korlát-relációk: #<, #>, #=<, #>=, #= #\= (mind xfx 700
operátorok)

Halmazkorlátok

• X in KTartomány , jelentése:X∈ H, aholH aKTartomány (konstans
tartomány) által leírt halmaz (Azin atom egyxfx 700 operátor);

• domain([ X, Y,...], Min , Max) : X∈ [Min,Max ], Y∈ [Min,Max ], . . .

Itt Min lehetSzámvagy inf (−∞), Max pedigSzámvagysup (+∞);
(Megjegyzés: a végtelen tartományok főleg kényelmi célokat szolgálnak: nem kell
kiszámolnunk az alsó/felső korlátokat, ha azok kikövetkeztethetők.)

Egy KTartomány a következ̋ok egyike lehet:

• felsorolás:{ Szám, ...} ,

• intervallum:( Min .. Max) , (xfx 550 operátor),

• metszet:KTartomány /\ KTartomány (yfx 500 , beépített op.),

• únió: KTartomány \/ KTartomány , (yfx 500 , beépített op.),

• komplemens:\ KTartomány , (fy 500 operátor).

Példák
| ?- X in (10..20)/\ (\{15}), Y in 6..sup, Z #= X+Y.

X in(10..14)\/(16..20), Y in 6..sup, Z in 16..sup ?

| ?- X in 10..20, X #\= 15, Y in {2}, Z #= X * Y.

Y = 2, X in(10..14)\/(16..20), Z in 20..40 ?

A térképszínezési feladat SICStus-ban

| ?- use_module(library(clpfd)).
...
| ?- domain([A,B,C,D,E], 1, 3),

A #> B, A #\= C, A #\= D, A #\= E,
B #\= C, B #\= D, C #\= E, D #< E.

A in 2..3, B in 1..2,
C in 1..3, D in 1..2, E in 2..3 ? ;
no

| ?- domain([A,B,C,D,E], 1, 3),
A #> B, A #\= C, A #\= D, A #\= E,
B #\= C, B #\= D, C #\= E, D #< E,
member(A, [1,2,3]). % címkézés, hivatalosan:

% indomain(A). % vagy:
% labeling([], [A]). % általánosan:
% labeling([], [A,B,C,D,E]).

A = 3, B = 2, C = 1, D = 1, E = 2 ? ;
no

| ?- domain([A,B,C,D,E], 1, 3),
A #> B, A #\= E, B #\= C, B #\= D, D #< E,

% A #\= C, A #\= E, C #\= E helyett:
all_distinct([A,C,E]).
% Az ,,A, C, E különböz őek’’ korlát okos
% megvalósítása, globális kombinatorikai korláttal

A = 3, B = 2, C = 1, D = 1, E = 2 ? ; no

Címkéző könyvtári eljárások — rövid el őzetes

• indomain( X) : X-et a tartománya által megengedett értékkel helyettesíti,
visszalépéskor felsorolja az összes értéket (növekedő sorrendben)

• labeling( Opciók , Változók ) : A Változók lista minden elemét
behelyettesíti, azOpciók lista által el̋oírt módon.

CSP/CLP programok: klasszikus példa

Kódaritmetikai feladat: SEND+MORE=MONEY

A feladvány: Írjon a betűk helyébe számjegyeket (azonosakhelyébe azonosakat,
különböz̋oek helyébe különbözőeket), úgy hogy az egyenlőség igaz legyen. Szám
elején nem lehet 0 számjegy.

send(SEND, MORE, MONEY) :-
length(List, 8),
domain(List, 0, 9), % tartományok
send(List, SEND, MORE, MONEY), % korlátok
labeling([], List). % címkézés

send(List, SEND, MORE, MONEY) :-
List= [S,E,N,D,M,O,R,Y],
alldiff(List), S #\= 0, M#\= 0,
SEND #= 1000* S+100* E+10* N+D,
MORE #= 1000* M+100* O+10* R+E,
MONEY #= 10000* M+1000* O+100* N+10* E+Y,
SEND+MORE #= MONEY.

% alldiff(L): L elemei mind különböz őek ( buta
% megvalósítás). Lényegében azonos a beépített
% all_different/1 kombinatorikai globális korláttal.
alldiff([]).
alldiff([X|Xs]) :- outof(X, Xs), alldiff(Xs).

outof(_, []).
outof(X, [Y|Ys]) :- X #\= Y, outof(X, Ys).

| ?- send(SEND, MORE, MONEY).
MORE = 1085, SEND = 9567, MONEY = 10652 ? ; no

| ?- List=[S,E,N,D,M,O,R,Y], domain(List, 0, 9),
send(List, SEND, MORE, MONEY).

List = [9,E,N,D,1,0,R,Y],
SEND in 9222..9866,
MORE in 1022..1088,
MONEY in 10244..10888,
E in 2..8, N in 2..8, D in 2..8,
R in 2..8, Y in 2..8 ? ; no

Szűkítési szintek

Informálisan, r(X,Y) bináris relációra

• Tartomány-szűkítés:X tartományából minden olyanx értéket elhagyunk,
amelyhez nem találhatóY tartományában olyany érték, amelyrer(x,y)
fennáll. Hasonlóan szűkítjükY tartományát. (Ez él-konzisztenciát eredményez.)

• Intervallum-szűkítési lépés:X tartományából elhagyjuk annakalsó vagy fels̋o
határát, ha ahhoz nem találhatóY tartományának széls̋o értékei közé es̋o
olyany érték, amelyrer(x,y) fennáll, és fordítva. Ezeket a lépéseket
ismételjük, ameddig szükséges.

Példa

• Legyen

– r(X,Y): X = abs(Y).

– X tartománya0..5

– Y tartománya{-1,1,3,4}

• A tartomány-szűkítés elhagyjaX tartományából a0,2,5 értékeket, eredménye
X∈ {1,3,4} .

• Az intervallum-szűkítésX tartományából csak az5 értéket hagyja el,
eredményeX∈ 0..4 .

• Az intervallum-szűkítés kétféle módon is gyengébb mint a tartomány-szűkítés:

– csak a tartomány szélső értékeit hajlandó elhagyni, ezért nem hagyja el a2
értéket;

– a másik változó tartományában nem veszi figyelembe a „lukakat”, így a
példábanY tartománya helyett annaklefedő intervallumát, azaz a-1..4
intervallumot tekinti — ezért nem hagyja elX-ből a0 értéket.

• Ugyanakkor az intervallum-szűkítés általában konstans idejű művelet, míg a
tartomány-szűkítés ideje (és az eredmény mérete) függ a tartományok méretétől.

Szűkítési szintek — definíciók

Jelölések

• LegyenC egyn-változós korlát,s egy tár,

• D(X, s) azX változó tartománya azs tárban,

• D′(X, s) = min(D(X, s))..max(D(X, s)), azX változó tartományátlefedő
(legszűkebb)intervallum.

A szűkítési szintek definíciója

• Tartomány-szűkítés (domain consistency)

C tartomány-szűkítő ha minden szűkítési lépés lefutása után az adottC korlát
él-konzisztens, azaz bármelyikXi változójához és annak tetszőleges
Vi ∈ D(Xi, s) megengedett értékéhez található a többi változónak olyan
Vj ∈ D(Xj, s) értéke (j = 1, . . . , i − 1, i + 1, . . . , n),hogyC(V1, . . . Vn)
fennálljon.

• Intervallum-szűkítés (interval consistency)

C intervallum-szűkít ő ha minden szűkítési lépés lefutása után igaz, hogyC
bármelyikXi változója esetén e változó tartományának mindkétvégpontjához
(azaz aVi = min(D(Xi, s)) illetve Vi = max(D(Xi, s)) értékekhez) található a
többi változónak olyanVj ∈ D′(Xj, s) értéke (j = 1, . . . , i − 1, i + 1, . . . , n),
hogyC(V1, . . . Vn) fennálljon.

Megjegyzések

• A tartomány-szűkítés lokálisan (egy korlátra nézve) a lehető legjobb;

• DE mégha minden korlát tartomány-szűkítő, a megoldás nem garantálható, pl.
| ?- domain([X,Y,Z], 1, 2), X#\=Y, X#\=Z, Y#\=Z.

• Egy CLP(FD) probléma megoldásának hatékonysága fokozható:

– több korlát összefogását jelentő ún. globális korlátokkal, pl.
all_distinct(L) : Az L lista csupa különböz̋o elemb̋ol áll;

– redundáns korlátok felvételével.

56



Garantált szűkítési szintek SICStusban

A SICStus által garantált szűkítési szintek

• A halmaz-korlátok (triviálisan) tartomány-szűkítők.

• A lineárisaritmetikai korlátok legalább intervallum-szűkítők.

• A nem-lineáris aritmetikai korlátokra nincs garantált sz˝ukítési szint.

• Ha egy változó valamelyik határa végtelen (inf vagysup ), akkor a változót
tartalmazó korlátokra nincs szűkítési garancia (bár az aritmetikai és
halmaz-korlátok ilyenkor is szűkítenek).

• A kés̋obb tárgyalandó korlátokra egyenként megadjuk majd a szűkítési szintet.

Példák

| ?- X in {4,9}, Y in {2,3}, Z #= X-Y.
% intervallum-szűkít ő:
X in {4}\/{9}, Y in 2..3, Z in 1..7 ?

| ?- X in {4,9}, Y in {2,3}, plus(Y, Z, X).
% plus(A, B, C): A+B=C tartomány-szűkít ő módon
X in {4}\/{9}, Y in 2..3, Z in(1..2)\/(6..7) ?

| ?- X in {4,9}, Y in {2}, / * azaz Y=2 * /, Z #= X-Y.
% tartomány-szűkít ő:
Y = 2, X in {4}\/{9}, Z in {2}\/{7} ?

| ?- X in {4,9}, Z #= X-Y, Y=2.
% így csak intervallum-szűkít ő!
% vö. fordítási idejű korlát-kifejtés
Y = 2, X in {4}\/{9}, Z in 2..7 ?

| ?-domain([X,Y], -10, 10), X * X+2* X+1 #= Y.
% Ez nem interv.-szűkít ő, Y<0 nem lehet!
X in -4..4, Y in -7..10 ?

| ?- domain([X,Y], -10, 10), (X+1) * (X+1) #= Y.
% garantáltan nem, de intervallum-szűkít ő:
X in -4..2, Y in 0..9 ?

Korlátok végrehajtása

A végrehajtás fázisai

• A korlát kifejtése elemi korlátokra (fordítási időben, lásd kés̋obb)
pl. X* X #> Y⇒ X* X #= Z, Z #> Y

• A korlát felvétele (posting):

– azonnali végrehajtás (pl.X #< 3), vagy

– démon létrehozása: első szűkítés elvégzése, újra-aktiválási feltételek
meghatározása, a démon elaltatása.

• A démon aktiválása

– szűkítés elvégzése,

– döntés a folytatásról:

∗ a démon lefut (ha a korlát már következménye a tárnak);
∗ vagy a démon újra elalszik.

Elemi korlátok működése — példák

A #\= B (tartomány-szűkít̋o)

• Mikor aktiválódik ? Ha vagyA vagyB konkrét értéket kap.

• Hogyanszűkít? A felvett értéket kihagyja a másik változó tartományából.

• Hogyanfolytatódik a démon végrehajtása? A démon befejezi működését
(lefut).

A #< B (tartomány-szűkít̋o)

• Aktiválás: haA alsó határa (minA) vagyB felső határa (maxB) változik

• Szűkítés: A tartományából kihagyja azX ≥ maxB értékeket,
B tartományából kihagyja azY ≤ min A értékeket

• Folytatás: ha maxA < min B, akkor lefut, különben újra elalszik. (SICStusban:
lefut, haA vagyB behelyettesít̋odik.)

Korlátok végrehajtása (folyt.)

all_distinct([A 1,...]) (tartomány-szűkít̋o)

• Aktiválás: ha bármelyik változó tartománya változik

• Szűkítés: (páros gráfokban maximális párosítást kereső algoritmus
segítségével) minden olyan értéket elhagy, amelyek eseténa korlát nem állhat
fenn. Példa:

| ?- A in 2..3, B in 2..3, C in 1..3,
all_distinct([A,B,C]).

C = 1, A in 2..3, B in 2..3 ?

• Folytatás: ha már csak egy nem-konstans argumentuma van, akkor lefut,
különben újra elalszik. (Jobb döntésnek tűnhet lefutni, ha a tartományok mind
diszjunktak, de a SICStus nem így csinálja, valószinűleg nem éri meg.)

X+Y #= T (intervallum-szűkít̋o)

• Aktiválás: ha bármelyik változó alsó vagy felső határa változik

• Szűkítés: T-t szűkiti a( min X+min Y)..( max X+max Y) intervallumra,
X-t szűkiti a( min T- max Y)..( max T- min Y) intervallumra,Y-t analóg
módon szűkiti.

• Folytatás: ha (a szűkités után) mindhárom változó konstans, akkor lefut
különben újra elalszik.

Példa a szűkítések kölcsönhatására

| ?- domain([X,Y], 0, 100), X+Y #=10, X-Y #=4.
X in 4..10, Y in 0..6 ?

| ?- domain([X,Y], 0, 100), X+Y #=10, X+2 * Y #=14.
X = 6, Y = 4 ?

Miért más a CLP(FD), mint a többi CLP rendszer?

A CLP könyvtárak összehasonlítása

clpq/r clpb clpfd

Korlátok: aritmetikai logikai aritmetikai.
logikai,
kombinatorikai,. . .

Egyszerű korlátok: lineárisak összes X in Halmaz

Összetett korlátok
végrehajtása:

várakozás, míg
lineáris nem lesz

nincs ilyen erősítés (szűkítés)

A tár
konzisztenciájának
biztosítása:

Gauss elimináció,
szimplex

Bináris Döntési
Diagrammok

triviális:
X in Halmaz→
Halmaz nem üres

Az összes korlát
konzisztenciájának
biztosítása:

lineáris esetben
automatikus

automatikus csak címkézésen
keresztül

Átlátszóság: fekete doboz fekete doboz üveg-doboz

Kiterjeszthet̋oség: nem nem igen

A CLP(FD) f ő jellemzői

• A tár konzisztenciájának biztosítása triviális.

• A lényeg a démonok erősít̋o (szűkít̋o) működésében van.

• A démonok nem látják egymást, csak a táron keresztül hatnak egymásra.

• Globális korlátok: egyszerre több (akárhány) korlátot helyettesítenek, így
erősebb szűkítést adnak (pl.all_distinct ).

• A megoldás megléte általában csak a címkézéskor derül ki.

A CLP(FD) jellemzői — példák

| ?- domain([X,Y,Z], 1, 2), X #\= Y, X #\= Z, Y #\= Z.
X in 1..2, Y in 1..2, Z in 1..2 ?

| ?- X #> Y, Y #> X.
Y in inf..sup, X in inf..sup ?

| ?- domain([X,Y], 1, 10), X #> Y, Y #> X.
no

| ?- statistics(runtime,_),
( domain([X,Y], 1, 1000000), X #> Y, Y #> X
; statistics(runtime,[_,T])
).

T = 3630 ?

A szűkítések nyomkövetése az FDBG könyvtár segítségével

| ?- use_module(library(fdbg)).
| ?- fdbg_on, fdbg_assign_name(X, x), fdbg_assign_name(Y , y),

domain([X,Y], 1, 10), X #> Y, Y #> X.

domain([<x>,<y>], ==> x = inf..sup -> 1..10,
1,10) y = inf..sup -> 1..10

Constraint exited.

<x> #>= <y>+1 ==> x = 1..10 -> 2..10, y = 1..10 -> 1..9

<x>+1 #=< <y> ==> x = 2..10 -> 2..8, y = 1..9 -> 3..9

<x> #>= <y>+1 ==> x = 2..8 -> 4..8, y = 3..9 -> 3..7

<x>+1 #=< <y> ==> x = 4..8 -> 4..6, y = 3..7 -> 5..7

<x> #>= <y>+1 ==> x = 4..6 -> {6}, y = 5..7 -> {5}
Constraint exited.

2 #=< 0 ==> Constraint failed.
% Valójában a korlát <x>+1 #=< <y>, azaz 6+1 #=< 5
no

Klasszikus CSP/CLP programok: a „zebra” feladat

A feladvány

Egy utcában öt különböző színű ház van egymás mellett. A házakban különböző
nemzetiségű és foglalkozású emberek laknak. Mindenki különböz̋o háziállatot tart
és más-más a kedvenc italuk is. A következőket tudjuk.

• Az angol a piros házban lakik.

• A festő japán.

• A norvég a balszéls̋o házban lakik.

• A zöld ház a fehérnek jobboldali
szomszédja.

• A diplomata a sárga házban lakik.

• A hegedűművész gyümölcslevet iszik.

• Az orvos szomszédja rókát tart.

• A spanyol kutyát tart.

• Az olasz a teát kedveli.

• A zöld házban lakó kávét iszik.

• A szobrász csigát tart.

• A tejet a középs̋o házban kedvelik.

• A norvég a kék ház mellett lakik.

• A diplomata melletti házban lovat
tartanak.

Kérdés: Kinek a háziállata a zebra?
(Lásd pl.http://brownbuffalo.sourceforge.net/zebra.html )

Modellezés

• változók meghatározása: egy-egy változó tartozik minden nemzetiséghez,
háziállathoz, házszinhez, foglalkozáshoz és italhoz.

• változóértékek kódolása: A változó értéke annak a háznak a száma (balról
számozva), amelynek lakóját, állatát, szinét, stb. jelöliaz adott változó.

• korlátok meghatározása:

– az egyes változó-csoportok mind különböznek:all_different/1
könyvtári korlát, pl.
all_different([Angol,Spanyol,Japán,Norvég,Olasz])

– két tulajdonság azonossága: egy#= korlát, pl. „Az angol a piros házban
lakik.” ⇒ Angol #= Piros

– két tulajdonság szomszédossága: házszámok különbsége 1, ill. 1 abszolút
értékű, pl. „A norvég a kék ház mellett lakik”⇒ abs(Norvég-Kék)#=1

– A sorban egy konkrét ház megnevezése: egy számmal való egyenlőség, pl.
„A tejet a középs̋o házban kedvelik.”⇒ Tej #= 3 .

62



A „zebra” feladvány CLPFD megoldása

:- use_module(library(lists)).
:- use_module(library(clpfd)).

% ZOwner a zebra tulajdonosának nemzetisége, All az
% összes változó értéke a "Kié a zebra" feladványban.
zebra(ZOwner, All):-

All = [England,Spain,Japan,Norway,Italy,
Dog, Zebra, Fox, Snail, Horse,
Green,Red,Yellow,Blue,White,
Painter,Diplomat,Violinist,Doctor,Sculptor,
Juice,Water,Tea,Coffee,Milk],

domain(All, 1, 5),
all_different([England,Spain,Japan,Norway,Italy]),
all_different([Green,Red,Yellow,Blue,White]),
all_different([Painter,Diplomat,Violinist,

Doctor,Sculptor]),
all_different([Dog,Zebra,Fox,Snail,Horse]),
all_different([Juice,Water,Tea,Coffee,Milk]),
England #= Red, Spain #= Dog,
Japan #= Painter, Italy #= Tea,
Norway #= 1, Green #= Coffee,
Green #= White+1, Sculptor #= Snail,
Diplomat #= Yellow, Milk #= 3,
Violinist #= Juice, nextto(Norway, Blue),
nextto(Fox, Doctor), nextto(Horse, Diplomat),
labeling([], All),
nth(N, [England,Spain,Japan,Norway,Italy], Zebra),
nth(N, [england,spain,japan,norway,italy], ZOwner).

% A és B szomszédos számok.
nextto(A, B) :- abs(A-B) #= 1.

| ?- zebra(ZOwner, All).
All = [3,4,5,1,2,4,5,1,3,2|...],
ZOwner = japan ? ; no

CSP/CLP programok: N vezér a sakktáblán

A feladvány

Egy N*N-es sakktáblán N vezért kell elhelyezni úgy, hogy egyik se üsse semelyik
másikat, azaz ne legyen két vezér ugyanabban a sorban, ugyanabban az oszlopban,
vagy ugyanazon átlós irányú vonal mentén.

Modellezés

• változók meghatározása: Minden vezérhöz egy változót rendelünk. AzXi

változó írja le azi. sorban lev̋o vezér helyzetét.

• változóértékek kódolása: AzXi változó azt az oszlopot jelöli, amelybe azi.
sorban lev̋o vezér kerül.

• korlátok meghatározása:

– ne legyen két vezér egy sorban: nem szükséges külön korlát, mert a
modellezés (változók jelentése) automatikusan biztosítja.

– ne legyen két vezér egy oszlopban:
Xi #\= X j, minden1 ≤ i < j ≤ N esetén.

– minden átlós vonalban legfeljebb egy vezér legyen: bármelykét vezér
vízszintes és függ̋oleges távolsága különbözzék: abs(X i-X j) #\= j − i,
minden1 ≤ i < j ≤ N esetén.

– Összegezve:mindenX, Y változópárra amelyek sortávolságaI (azazX =
Xi, Y = Xj, I = abs(i − j)) a következ̋o három korlát fennállását kell
biztosítani:
Y #\= X, Y #\= X-I, Y #\= X+I

– A fenti korlátok eljárásba foglalása:

% Az X és Y oszlopokban I sortávolságra lev ő
% vezérek nem támadják egymást.
no_threat(X, Y, I) :-

Y #\= X, Y #\= X-I, Y #\= X+I.

Egy bonyolultabb példa: mágikus sorozatok

Definíció: EgyL = (x0, . . . ,xn−1) sorozatmágikus(xi ∈ [0..n− 1]), haL-ben azi
szám pontosanxi-szer fordul el̋o (mindeni ∈ [0..n − 1]-re).

Példa: n=4 esetén (1,2,1,0) és (2,0,2,0) mágikus sorozatok.

% Az L lista egy N hosszúságú mágikus sorozat.
magikus(N, L) :-

length(L, N), N1 is N-1, domain(L, 0, N1),
elofordulasok(L, 0, L),
labeling([], L). % most felesleges

% elofordulasok([E i , E i+1 , ...], i, Sor): Sor-ban az i
% szám Ei -szer, az i+1 szám E i+1 -szer stb. fordul el ő.
elofordulasok([], _, _).
elofordulasok([E|Ek], I, Sor) :-

pontosan(I, Sor, E),
J is I+1, elofordulasok(Ek, J, Sor).

% pontosan(I, L, E): Az I szám L-ben E-szer fordul el ő.
pontosan(I, L, 0) :- outof(I, L).
pontosan(I, [I|L], N) :-

N #> 0, N1 #= N-1, pontosan(I, L, N1).
pontosan(I, [X|L], N) :-

N #> 0, X #\= I, pontosan(I, L, N).

Példafutás:
| ?- spy pontosan/3, magikus(4, L).

+ 1 1 Call: pontosan(0,[_A,_B,_C,_D],_A) ? s
?+ 1 1 Exit: pontosan(0,[1,0,_C,_D],1) ? z

+ 2 1 Call: pontosan(1,[1,0,_C,_D],0) ? s
+ 2 1 Fail: pontosan(1,[1,0,_C,_D],0) ? z
+ 1 1 Redo: pontosan(0,[1,0,_C,_D],1) ? s

?+ 1 1 Exit: pontosan(0,[2,0,0,_D],2) ? z
(...)

+ 4 1 Call: pontosan(2,[2,0,0,_D],0) ? s
+ 4 1 Fail: pontosan(2,[2,0,0,_D],0) ? z

(...)
?+ 1 1 Exit: pontosan(0,[3,0,0,0],3) ? z
(...)
?+ 1 1 Exit: pontosan(0,[2,0,_D,0],2) ?

Mágikus sorozatok: redundáns korlátok

Állítás: Ha azL = (x0, . . . ,xn−1) sorozat mágikus,
akkor

∑
i<n xi = n, és

∑
i<n i ∗ xi = n.

Hatékonyabb változat, a fenti redundáns korlátokkal

% N=10 esetén kb. 50-szer gyorsabb az el őz ő programnál!
magikus2(N, L) :-

length(L, N), N1 is N-1, domain(L, 0, N1),
osszege(L, S), %

∑
i∈ [1..N] Li = S

szorzatosszege(L, 0, SP), %
∑

i∈ [0..N-1 ] i ∗ Li+1 = SP
call(S #= N), call(SP #= N), % lásd a megjegyzést
elofordulasok(L, 0, L). % lásd az el őz ő lapon

Megjegyzés

• Az aritmetikai beépített eljárások megengednek (aritmetikai) struktúrákat
tartalmazó változókat, pl.Kif = S1+S2, ..., Kif =:= 0 .

• CLPFD-ben ez nem megengedett:Kif=S1+S2, ..., Kif #= 0 ⇒
Hiba! Ennek oka: a korlát-kifejtés csak betöltéskor történik meg.

• A megoldás a korlát-kifejtési fázis késleltetése:Kif=S1+S2, ...,
call(Kif #= 0) .

Segédeljárások
% osszege(L, Ossz): Ossz =

∑
i Li

osszege([], 0).
osszege([X|L], X+S) :- osszege(L, S).

% szorzatosszege(L, I, Ossz): Ossz = I ∗ L1 + (I+1) ∗ L2 + . . .
szorzatosszege([], _, 0).
szorzatosszege([X|L], I, I * X+S) :-

J is I+1, szorzatosszege(L, J, S).

| ?- magikus2(4, L).
% visszalépés nélkül adja ki az els ő megoldást!

+ 1 1 Call: pontosan(0,[_A,_B,_C,_D],_A) ?
(...)
?+ 1 1 Exit: pontosan(0,[2,0,2,0],2) ? z

Reifikáció: korlátok tükrözése

Egy korlát tükrözése (reifikációja):

• a korlát igazságértékének „tükrözése” egy 0-1 értékű korlát-változóban;

• jelölése:C #<=> B, jelentése:B tartománya0..1 ésB csakkor1, haC igaz;

• példa:(X #>= 3) #<=> B jelentése:B azX ≥ 3 egyenl̋oség
igazságértéke.

Megjegyzések

• Az ún. formula-korlátok (az eddig ismertetett aritmetikaiés halmaz-korlátok)
mind tükrözhet̋oek.

• A globális korlátok (pl.all_different/1, all_distinct/1 ) nem
tükrözhet̋oek.

• A tükrözött korlátok is „közönséges” korlátok, csak definíciójuk és
végrehajtásuk módja speciális.

• Példa: a0..5 tartományon a(X #>= 3) #<=> B korlát teljesen
megegyezik aB #= X/3 korláttal.

Tükrözött korlátok végrehajtása

• A C <=> B tükrözött korlát végrehajtása többféle szűkítést igényel:

a. amikorB-ről kiderül valami (azaz behelyettesítődik): haB=1, fel kell venni
(post) a korlátot, haB=0, fel kell venni a negáltját.

b. amikorC-ről kiderül, hogy levezethető a tárból:B=1 kell legyen

c. amikor¬C-ről kiderül, hogy levezethető a tárból:B=0 kell legyen

• A fenti a., b. és c. szűkítések elvégzését három különböző démon végzi.

• A levezethet̋oség-vizsgálat (b. és c.) különböző „ambíciókkal”, különböz̋o
bonyolultsági szinteken végezhető el.

Reifikáció — példák

• Alappélda, csakB szűkül:

| ?- X#>3 #<=> B. ⇒ B in 0..1

• HaB értéket kap, akkor a rendszer felveszi a korlátot ill. a negáltját:

| ?- X#>3 #<=> B, B = 1. ⇒ X in 4..sup
| ?- X#>3 #<=> B, B = 0. ⇒ X in inf..3

• Ha levezethet̋o a korlát vagy negáltja, akkorB értéket kap.

| ?- X#>3 #<=> B, X in 15..sup. ⇒ B = 1
| ?- X#>3 #<=> B, X in inf..0. ⇒ B = 0

• Ha a tár megengedi a korlát és negáltja teljesülését is, akkor B nem kap értéket.

| ?- X#>3 #<=> B, X in 3..4. ⇒ B in 0..1

• A rendszer kikövetkezteti, hogy az adott tárbanX ésY távolsága legalább1:

| ?- abs(X-Y)#>1 #<=> B, X in 1..4, Y in 6..10.
⇒ B = 1

• Bár a távolság-feltétel itt is fennáll, a rendszer nem vesziészre!

| ?- abs(X-Y)#>1 #<=> B, X in {1,5}, Y in {3,7}.
⇒ B in 0..1

• Ennek itt az az oka, hogy az aritmetika nem tartomány-konzisztens.

| ?- D #= X-Y,
AD #= abs(D), AD#>1 #<=> B,
X in {1,5}, Y in {3,7}.

⇒ D in -6..2, AD in 0..6, B in 0..1

| ?- plus(Y, D, X), ⇐ tartomány-konzisztens összegkorlát
AD #= abs(D), AD#>1 #<=> B,
X in {1,5}, Y in {3,7}.

⇒ D in {-6,-2,2}, AD in {2,6}, B = 1

Korlátok levezethetősége

A levezethet̋oség (entailment) felderítésének szintjei

• Tartomány-levezethetőség (domain-entailment):
A C n-változós korláttartomány-levezethet̋o azs tárból, ha változóinaks-ben
megengedett tetszőlegesVj ∈ D(Xj, s) értékkombinációjára (j = 1, . . . , n),
C(V1, . . . Vn) fennáll.

• Intervallum-levezethetőség (interval-entailment):
C intervallum-levezethet̋o s-ből, ha mindenVj ∈ D′(Xj, s)
értékkombinációra (j = 1, . . . , n), C(V1, . . . Vn) fennáll.

Megjegyzések

• HaC intervallum-levezethető, akkor tartomány-levezethető is.

• A tartomány-levezethetőség vizsgálata általában bonyolultabb, mint az
intervallum-levezethetőségé. Például aX #\= Y korlát:

– tartomány-levezethető, haX ésY tartományai diszjunktak (a tartomány
méretével arányos költség) ;

– intervallum-levezethető, haX ésY tartományainak lefed̋o intervallumai
diszjunktak (konstans költség).

A SICStus által garantált levezethet̋oségi szintek

• A tükrözött halmaz-korlátok kiderítik a tartomány-levezethet̋oséget.

• A tükrözött lineáris aritmetikai korlátok legalább az
intervallum-levezethetőséget kiderítik.

• A tükrözött nem-lineáris aritmetikai korlátokra nincs garantált szint.

Példák

| ?- X in 1..4, X #< Y #<=> B, X+Y #=9.
B = 1, X in 1..4, Y in 5..8 ?

| ?- X+Y #= Z #<=> B, X=1, Z=6, Y in 1..10, Y#\=5.
X = 1, Z = 6, Y in(1..4)\/(6..10), B in 0..1 ?
% X+Y #\= Z tartomány-, de nem interv.-levezethet ő!

Mágikus sorozatok (folyt.)

Tükrözést használó változat

magikus3(N, L) :-
length(L, N),
N1 is N-1, domain(L, 0, N1),
osszege(L, S), call(S #= N),
szorzatosszege(L, 0, SS), call(SS #= N),
elofordulasok3(L, 0, L),
labeling([], L). % most már kell a címkézés!

% A korábbi elofordulasok/3 másolata
elofordulasok3([], _, _).
elofordulasok3([E|Ek], I, Sor) :-

pontosan3(I, Sor, E),
J is I+1, elofordulasok3(Ek, J, Sor).

% pontosan3(I, L, E): L-ben az I E-szer fordul el ő.
pontosan3(_, [], 0).
pontosan3(I, [X|L], N) :-

X #= I #<=> B, N #= N1+B, pontosan3(I, L, N1).

A mágikus sorozat megoldásainak összehasonlítása
Az összes megoldás előállítási ideje másodpercben, 1 perc időkorláttal, Pentium III,
600 MHz processzoron („—” = id̋otúllépés).

variáns/adat n=10 n=20 n=40 n=80 n=160 n=320
választós 13.90 — — — — —
választós+osszege 0.22 — — — — —
vál.+szorzatosszege 0.02 0.55 44.04 — — —
vál.+ossz +szorzossz 0.02 0.29 17.98 — — —
tükrözéses 0.05 1.07 24.02 — — —
tükrözéses+osszege 0.01 0.14 1.71 20.15 — —
tükr.+szorzatosszege 0.01 0.04 0.18 0.94 4.75 25.77
tükr.+ossz +szorzossz 0.01 0.05 0.19 0.95 4.61 23.57

Logikai korlátok

Logikai korlát argumentuma lehet

• egyB változó,B automatikusan a0..1 tartományra szűkül;

• egy tetsz̋oleges tükrözhető aritmetikai- vagy halmazkorlát;

• egy tetsz̋oleges logikai korlát.

A logikai korlátok (egyben függvényjelként is használhatók)

#\ Q negáció op(710, fy, #\).
P #/\ Q konjunkció op(720, yfx, #/\).
P #\ Q kizáró vagy op(730, yfx, #\).
P #\/ Q diszjunkció op(740, yfx, #\/).
P #=> Q implikáció op(750, xfy, #=>).
Q #<= P implikáció op(750, yfx, #<=).
P #<=> Q ekvivalencia op(760, yfx, #<=>).

A tükrözött és logikai korlátok kapcsolata

• A korábban bevezetett tükrözési jelölés (C <=> B) a fenti
logikaikorlát-fogalom speciális esete.

• De: a (C <=> B) alakúelemikorlát az, amire a logikai korlátok
visszavezet̋odnek.

• Példa:X#=4 #\/ Y#>6 −→
X#=4#<=>B1, Y#>6#<=>B2, B1+B2 #>0

• Vigyázat! A diszjunktív logikai korlátok gyengén szűkítenek, pl. egy n-tagú
diszjunkció csak akkor tud szűkíteni, ha egy kivételével valamennyi tagjának a
negáltja levezethetőve válik (a példában haX#\=4 vagyY#=<6 levezethet̋o
lesz).

Példa: lovagok, lóköt̋ok és normálisak

Egy szigeten minden bennszülött lovag, lókötő, vagy normális. A lovagok mindig
igazat mondanak, a lókötők mindig hazudnak, a normális emberek pedig néha
hazudnak, néha igazat mondanak. Kódolás:normális → 2, lovag → 1,
lóköt ő → 0.

:- use_module(library(clpfd)).
:- op(700, fy, nem). :- op(900, yfx, vagy).
:- op(800, yfx, és). :- op(950, xfy, mondja).

% A B bennszülött mondhatja az Áll állítást.
B mondja Áll :- értéke(B mondja Áll, 1).

% értéke(A, Érték): Az A állítás igazságértéke Érték.
értéke(X = Y, E) :-

X in 0..2, Y in 0..2, E #<=> (X #= Y).
értéke(X mondja M, E) :-

X in 0..2, értéke(M, E0),
E #<=> (X #= 2 #\/ E0 #= X).

értéke(M1 és M2, E) :-
értéke(M1, E1), értéke(M2, E2), E #<=> E1 #/\ E2.

értéke(M1 vagy M2, E) :-
értéke(M1, E1), értéke(M2, E2), E #<=> E1 #\/ E2.

értéke(nem M, E) :-
értéke(M, E0), E #<=> #\E0.

% http://www.math.wayne.edu/~boehm/Probweek2w99sol.h tm
% We are given three people, A, B, C, one of whom is
% a knight, one a knave, and one a normal (but not
% necessarily in that order). They make the following
% statements. A: I am normal
% B: A is right
% C: I am not normal
| ?- all_different([A,B,C]), A mondja A = 2,

B mondja A = 2, C mondja nem C =2,
labeling([], [A,B,C]).

A = 0, B = 2, C = 1 ? ; no

Globális aritmetikai korlátok

Ezek a korlátok nem tükrözhetőek.

scalar_product(Coeffs, Xs, Relop , Value [,Options] )
Igaz, ha aCoeffs ésXs listák skalárszorzata aRelop relációban van aValue
értékkel, aholRelop aritmetikai összehasonlító operátor (#=, #< , stb.).
Alapértelmezésben intervallum-szűkítést biztosít, kivéve haOptions =
[consistency(domain)] , amikor is tartomány-konzisztensen szűkít.
Coeffs egészekb̋ol álló lista,Xs elemei ésValue egészek vagy korlát változók
lehetnek.
Megjegyzés: minden lineáris aritmetikai korlát átalakítható egy
scalar_product hívássá.

sum(Xs, Relop , Value) Jelentése:Σ Xs Relop Value .
Ekvivalens a következ̋ovel: scalar_product(Csupa1, Xs, Relop ,
Value) , aholCsupa1 csupa1 számból álló lista,Xs-sel azonos hosszú.

knapsack(Coeffs, Xs, Value)
JelentéseCoeffs ésXs skalárszorzataValue . Tartomány-konzisztenciát biztosít.
Feltétel: Csak nem-negatív számok megengedettek, a változók véges tartományúak
kell legyenek.

minimum(Value, Xs), maximum(Value, Xs)
Jelentése: azXs lista elemeinek minimuma/maximumaValue .

Példa
send(List, SEND, MORE, MONEY) :-

List= [S,E,N,D,M,O,R,Y],
Pow10 = [1000,100,10,1],
all_different(List), S #\= 0, M#\= 0,
scalar_product(Pow10, [S,E,N,D], #=, SEND),
% SEND #= 1000* S+100* E+10* N+D,
scalar_product(Pow10, [M,O,R,E], #=, MORE),
% MORE #= 1000* M+100* O+10* R+E,
scalar_product([10000|Pow10], [M,O,N,E,Y],

#=, MONEY),
% MONEY #= 10000* M+1000* O+100* N+10* E+Y,
SEND+MORE #= MONEY.

Ezzel befejeztük a halmaz-, aritmetikai, logikai és tükrözött korlátok
ismertetését.

A formula-korlátok megvalósítása

Formula-korlátok

• Formula-korlátnak hívjuk az operátoros jelöléssel írt korlátot, azaz az eddig
ismertetetteket, kivéve a globális aritmetikai korlátokat.

• A formula-korlátokat a rendszer nem könyvtári eljárással valósítja meg, hanem
a Prologgoal_expansion/5 kampójának segítségével.

• A kampó-eljárásfordítási időbena formula-korlátot, egy
scalar_product/4 korlátra, és/vagy nem-publikus elemi korlátokra fejti ki.

• A formula-korlátok kifejtésecall/1 -be ágyazással elhalasztható a korlát
futási időbenvaló felvételéig.

A legfontosabb elemi korlátok aclpfd modulban

• aritmetika:’x+y=t’/3 ’x * y=z’/3 ’x/y=z’/3 ’x mod y=z’/3
’|x|=y’/2 ’max(x,y)=z’/3 ’min(x,y)=z’/3

• összehasonlítás:’x=y’/2, ’x=<y’/2, ’x\\=y’/2 és tükrözött
változataik:’x Rel y’(X,Y,B) , aholRel ∈ { = =< \= } .

• halmaz-korlátok:propagate_interval(X,Min,Max)
prune_and_propagate(X,Halmaz)

• logikai korlátok:
bool(Muvkod,X,Y,Z) % jelentése: X Muv Y = Z

• optimalizálások:’x * x=y’/2 ’ax=t’/3 ’ax+y=t’/4 ’ax+by=t’/5
’t+u=<c’/3 ’t=u+c’/3 ’t=<u+c’/3 ’t\\=u+c’/3 ’t>=c’/2
stb.

Az elemi korlátok szűkítési szintje

• Definíció: A C korlátpont-szűkítő, ha minden olyan tár esetén
tartomány-szűkít̋o, amelybenC változói, legfeljebb egy kivételével be vannak
helyettesítve. (Másképpen: ha minden ilyen tár esetén a korlát a
behelyettesítetlen változót pontosan aC reláció által megengedett értékekre
szűkíti.)

• Az elemi korlátok többsége pont-szűkítő (kivétel: mod).

Korlátok kifejtése

Példák (clpfd betöltése után)

| ?- use_module(library(clpfd)).
| ?- clpfd:goal_expansion(X * X+2* X+1 #= Y, _, user, G, []).

G = clpfd:(’x * x=y’(X,_A),
scalar_product([1,-2,-1],[Y,X,_A],#=,1)) ?

| ?- clpfd:goal_expansion((X+1) * (X+1) #= Y, _, user, G, []).
G = clpfd:(’t=u+c’(_A,X,1),’x * x=y’(_A,Y)) ?

| ?- clpfd:goal_expansion(abs(X-Y)#>1, _, user, G, []).
G = clpfd:(’x+y=t’(Y,_A,X),

’|x|=y’(_A,_B),’t>=c’(_B,2)) ?

| ?- clpfd:goal_expansion(X#=4 #\/ Y#>6, _, user, G, []).
G = clpfd:’x=y’(X,4,_A),

clpfd:’x=<y’(7,Y,_B),
clpfd:bool(3,_A,_B,1) ? % 3 a \/ kódja

| ?- clpfd:goal_expansion(X * X* X* X #= 16, _, user, G, []).
G = clpfd:(’x * x=y’(X,_A),’x * y=z’(_A,X,_B),

’x * y=z’(_B,X,16)) ?

| ?- clpfd:goal_expansion(X in {1,2}, _, user, G, []).
G = clpfd:propagate_interval(X,1,2) ?

| ?- clpfd:goal_expansion(X in {1,2,5}, _, user, G, []).
G = clpfd:prune_and_propagate(X,[[1|2],[5|5]]) ?

Megjegyzések

• Lineáris korlátok esetén a kifejtés megőrzi a pont- és intervallum-szűkítést.

• Általános esetben a kifejtés még a pont-szűkítést semőrzi meg, pl
| ?- X in 0..10, X * X* X* X#=16. −→ X in 1..4

CLPFD segédeljárások

Statisztika

• fd_statistics(Kulcs, Érték) : A Kulcs -hoz tartozó számláló
Érték -ét kiadja és lenullázza. Lehetséges kulcsok és számlált események:

– constraints — korlát létrehozása;

– resumptions — korlát felébresztése;

– entailments — korlát (vagy negáltja) levezethetővé válásának észlelése;

– prunings — tartomány szűkítése;

– backtracks — a tár ellentmondásossá válása (Prolog meghiúsulások
nem számítanak).

• fd_statistics : az összes számláló állását kiírja és lenullázzaőket.

% Az N-vezér feladat összes megoldása Ss, Lab címkézéssel va ló
% végrehajtása Time msec-ig tart és Btrks FD visszalépést ig ényel.
run_queens(Lab, N, Ss, Time, Btrks) :-

fd_statistics(backtracks, _), statistics(runtime, _),
findall(Q, queens(Lab, N, Q), Ss),
statistics(runtime, [_,Time]),
fd_statistics(backtracks, Btrks).

Válaszok formája (a még le nem futott, alvó korlátok kiírásaa válaszban)

• clpfd:full_answer : ez egy dinamikus kampó eljárás. Alaphelyzetben
nincs egy klóza sem, tehát nem sikerül. Ez esetben a rendszeregy kérdésre való
válaszoláskor csak a kérdésben előforduló változók tartományát írja ki, az alvó
korlátokat nem. Ha felveszünk egy ilyen eljárást és az sikeresen fut le, akkor a
válaszban az összes változó mellett kiírja még a le nem futott összes korlátot is.

| ?- domain([X,Y], 1, 10), X+Y#=5. ⇒ X in 1..4, Y in 1..4 ?
| ?- assert(clpfd:full_answer). ⇒ yes
| ?- domain([X,Y], 1, 10), X+Y#=5. ⇒ clpfd:’t+u=c’(X,Y,5),

X in 1..4, Y in 1..4 ?
| ?- X+Y #= Z #<=> B. ⇒ clpfd:’t=u IND’(Z,_A)#<=>B,

clpfd:’x+y=t’(X,Y,_A), B in 0..1, ...
| ?- retract(clpfd:full_answer). ⇒ yes
| ?- X+Y #= Z #<=> B. ⇒ B in 0..1, ...

CLPFD segédeljárások (folyt.)

FD változók bels̋o jellemzői

• Az FD változókról a könyvtár által tárolt információk lekérdezhet̋ok.

• Ezek felhasználhatók a címkézésben, globális korlátok írásában ill.
nyomkövetésben.

• Vigyázat! Félreértés veszélye! Minden más használat nagy eséllyel hibás.

FD változók felismerése

• fd_var(V) : V egy a clpfd könyvtár által ismert változó.

Tartományok pillanatnyi jellemz őinek lekérdezése

• fd_min(X, Min) : A Min paramétert egyesíti azX változó tartományának
alsó határával (ez egy szám vagyinf lehet).

• fd_max(X, Max) : Max azX felső határa (szám vagysup ).

• fd_size(X, Size) : Size azX tartományának számossága (szám vagy
sup ).

• fd_dom(X, Range) : Range azX változó tartománya,
KonstansTartomány formában

• fd_set(X, Set) : Set azX tartománya ún. FD-halmaz formában.

• fd_degree(X, D) : DazX-hez kapcsolódó korlátok száma.

Példák

| ?- X in (1..5)\/{9}, fd_min(X, Min), fd_max(X, Max),
fd_size(X, Size).

Min = 1, Max = 9, Size = 6, X in(1..5)\/{9} ?
| ?- X in (1..9)/\ \(6..8), fd_dom(X, Dom), fd_set(X, Set).

Dom = (1..5)\/{9}, Set = [[1|5],[9|9]], X in ... ?
| ?- queens_nolab(8, [X|_]), fd_degree(X, Deg).

Deg = 21, X in 1..8 ? % 21 = 7 * 3

FD-halmazok

Az FD-halmaz fogalma, alapműveletei

• Az FD-halmaz formátum a tartományok belső ábrázolási formája.

• Absztrakt adattípusként használandó, alapműveletei:

– is_fdset(S) : S egy korrekt FD-halmaz.

– empty_fdset(S) : S az üres FD-halmaz.

– fdset_parts(S, Min, Max, Rest) : Az S FD-halmaz áll egy
Min..Max kezd̋o intervallumból és egyRest maradék FD-halmazból,
aholRest minden eleme nagyobbMax+1-nél. Egyaránt használható
FD-halmaz szétszedésére és építésére.

| ?- X in (1..9) /\ \(6..8), fd_set(X, _S),
fdset_parts(_S, Min1, Max1, _).

Min1 = 1,
Max1 = 5,
X in(1..5)\/{9} ?

• Az FD-halmaz tényleges ábrázolása:[Alsó|Fels ő] alakú szeparált zárt
intervallumok rendezett listája. (A ‘.(_,_) ’ struktúra memóriaigénye
33%-kal kevesebb mint bármely más ‘f(_,_) ’ struktúráé.)

| ?- X in (1..9) /\ \(6..8), fd_set(X, S).
S = [[1|5],[9|9]],
X in(1..5)\/{9} ?

• FD-halmaz is használató szűkítésre:

– X in_set Set : Az X változót aSet FD-halmazzal szűkíti.

– Vigyázat! Ha a korlát-felvételi fázisban egy változó tartományát egymásik
tartományának függvényében szűkítünk, ezzel nem érhetünk el „démoni”
szűkít̋o hatást, hiszen ez a szűkítés csakegyszerfut le. Az in_set eljárást
csak globális korlátok ill. testreszabott címkézés megvalósítására célszerű
használni.

FD-halmazok (folyt.)

FD-halmazokat kezel̋o további eljárások

• fdset_singleton(Set, Elt) : Set az egyetlenElt -ből áll.

• fdset_interval(Set, Min, Max) : Set aMin..Max intervallum
(oda-vissza használható).

• empty_interval(Min, Max) : Min..Max egy üres intervallum.
Ekvivalens a\+fdset_interval(_, Min, Max) hívással.

• fdset_union(Set1, Set2, Union) : Set1 ésSet2 úniójaUnion ,
fdset_union(ListOfSets, Union) : aListOfSets lista elemeinek
úniójaUnion .

• fdset_intersection/[3,2] : Két halmaz ill. egy listában megadott
halmazok metszete.

• fdset_complement/2 : Egy halmaz komplemense.

• fdset_member(Elt, Set) : Elt eleme aSet FD-halmaznak.

• list_to_fdset(List, Set), fdset_to_list(Set, List) :
Számlista átalakítása halmazzá, és fordítva.

• range_to_fdset(Range, Set),
fdset_to_range(Set, Range) :
Konstans tartomány átalakítása halmazzá és viszont.

Példa

| ?- list_to_fdset([2,3,5,7], _FS1),
fdset_complement(_FS1, _FS2),

% _FS2 ↔ \{2,3,5,7}
fdset_interval(_FS3, 0, sup),

% _FS3 ↔ 0..sup
fdset_intersection(_FS2, _FS3, FS),

% FS ↔ (0..sup)/\ \{2,3,5,7}
fdset_to_range(FS, Range),
X in_set FS.

FS = [[0|1],[4|4],[6|6],[8|sup]],
Range = (0..1)\/{4}\/{6}\/(8..sup),
X in(0..1)\/{4}\/{6}\/(8..sup) ?

Címkézési (keresési) stratégiák

CSP programok szerkezete (ismétlés!)

• változók és tartományaik megadása,

• korlátok felvétele (lehetőleg választási pontok létrehozása nélkül),

• címkézés (keresés).

A címkézési fázis feladata

• Adott változók egy halmaza,

• ezeket a tartományaik által megengedett értékekre szisztematikusan be kell
helyettesíteni

• (miközben a korlátok fel-felébrednek, és visszalépést okoznak a nem
megengedett állapotokban).

• Mindezt a lehet̋o leggyorsabban, a lehető legkevesebb visszalépéssel kell
megoldani.

A keresés célja lehet

• egyetlen(tetsz̋oleges) megoldás előállítása,

• azösszesmegoldás el̋oállítása,

• a valamilyen szempontbóllegjobb megoldás el̋oállítása.

A keresési stratégia paraméterezési lehetőségei

• Milyen sorrendben kezeljük az egyes változókat?

• Milyen választási pontot hozunk létre?

• Milyen irányban járjuk be a változó tartományát?

Keresési stratégiák — példák

Hogyan függ a keresési tér a változó-sorrendtől?

• | ?- X in 1..4, Y in 1..2,
indomain(X),
indomain(Y).

Y

11 12XY 21 22 31 32 41 42

X

• | ?- X in 1..4, Y in 1..2,
indomain(Y),
indomain(X).

X

Y

11 21 31 41 12 22 32 42XY

• A first-fail elv: a kisebb tartományú változót előbb címkézzük —
kevesebb választási pont, remélhetően kisebb keresési tér.

• Példa feladatspecifikus sorrendre: az N vezér feladatban érdemes a középső
sorokba tenni le először a vezéreket, mert ezek a többi változó tartományát
jobban megszűrik, mint a szélsőkbe tettek.

Milyen szerkezetű keresési tereket hozhatunk létre?

• felsorolás:| ?- X in 1..4,
labeling([enum], [X]).

1 2 3 4

• kettévágás:| ?- X in 1..4,
labeling([bisect], [X]).

1 2 3 4

>2=<2

• lépegetés:| ?- X in 1..4,
labeling([step], [X]).

43

> 1

> 2

> 3

1

2

Címkéző eljárások

A címkézés alap-eljárása:labeling(Opciók, VáltozóLista)
A VáltozóLista minden elemét minden lehetséges módon behelyettesíti, az
Opciók lista által el̋oírt módon. Az alábbi csoportok mindegyikéből legfeljebb egy
opció szerepelhet. Hibát jelez, ha aVáltozóLista -ban van nem korlátos
tartományú változó. Ha az első négy csoport valamelyikéből nem szerepel opció,
akkor adőlt betűvel szedett alapértelmezés lép életbe.

1. a változó kiválasztása:leftmost , min, max, ff, ffc, variable(Sel)

2. a választási pont fajtája:step , enum, bisect, value(Enum)

3. a bejárási irány:up , down

4. a keresett megoldások:all , minimize(X), maximize(X)

5. a gyűjtend̋o statisztikai adat:assumptions(A)

6. a balszéls̋o ágtól való eltérés korlátozása:discrepancy(D)

7. időkorlát: time_out(MSec,Result)

A címkézés menete

a. Ha a változólista üres, akkor a címkézés sikeresen véget ér.Egyébként
kiválasztunk bel̋ole egyX elemet az 1. csoportbeli opció által előírt módon.

b. HaX behelyettesített, akkor a változólistából elhagyjuk, és az a. pontra megyünk.

c. Egyébként azX változó tartományát felosztjuk két vagy több diszjunkt részre a 2.
csoportbeli opció szerint (kivévevalue(Enum) esetén, amikor is azonnal az
e. pontra megyünk).

d. A tartományokat elrendezzük a 3. csoportbeli opció szerint.

e. Létrehozunk egy választási pontot, amelynek ágain sorra leszűkítjük azX változót
a kiválasztott tartományokra.

f. Minden egyes ágon azX szűkítése értelemszerűen kiváltja a rá vonatkozó korlátok
felébredését. Ha ez meghiúsulást okoz, akkor visszalépünkaze. pontra és ott a
következ̋o ágon folytatjuk.

g. Ha X most már behelyettesített, akkor elhagyjuk a változólistából. Ezután
mindenképpen folytatjuk aza. pontnál.

h. Eközben értelemszerűen követjük a 4.-7. csoportbeli opciók előírásait is.

Speciális címkézési eljárás:indomain(X)

Ekvivalens alabeling([enum], [X]) hívással.

A címkézés menete — példa
• A példa:

X in 1..3, Y in 1..2, X#>=Y, labeling([min], [X,Y]).

• A min opció a legkisebb alsó határú változó kiválasztását írja elő.

| ?- fdbg_assign_name(X, x), fdbg_assign_name(Y, y),
X in 1..3, Y in 1..2, X #>= Y, fdbg_on,
labeling([min], [X,Y]).

% The clp(fd) debugger is switched on
Labeling [1, <x>]: starting in range 1..3.
Labeling [1, <x>]: step: <x> = 1

<y>#=<1 y = 1..2 -> {1} Constraint exited.
X = 1, Y = 1 ? ;

Labeling [1, <x>]: step: <x> >= 2
<y>#=<<x> y = 1..2, x = 2..3 Constraint exited.
Labeling [6, <y>]: starting in range 1..2.
Labeling [6, <y>]: step: <y> = 1

Labeling [8, <x>]: starting in range 2..3.
Labeling [8, <x>]: step: <x> = 2

X = 2, Y = 1 ? ;
Labeling [8, <x>]: step: <x> >= 3

X = 3, Y = 1 ? ;
Labeling [8, <x>]: failed.

Labeling [6, <y>]: step: <y> >= 2
Labeling [12, <x>]: starting in range 2..3.
Labeling [12, <x>]: step: <x> = 2

X = 2, Y = 2 ? ;
Labeling [12, <x>]: step: <x> >= 3

X = 3, Y = 2 ? ;
Labeling [12, <x>]: failed.

Labeling [6, <y>]: failed.
Labeling [1, <x>]: failed.

A keresési fa

X=1

X>=3

X>= 2

X=2

Y>= 2Y=1

X>=3X=2

<3,2><2,2><3,1><2,1>

X#>=Y

Y=1

<1,1>

Címkézési opciók

A címkézend̋o változó
A következ̋o címkézend̋o változó kiválasztási szempontjai (ahol több szempont van,
a kés̋obbi csak akkor számít, ha a megelőző szempont(ok) szerint több azonos
értékű van):

• leftmost (alapértelmezés) — legbaloldalibb;

• min — a legkisebb alsó határú; ha több ilyen van, közülük a legbaloldalibb;

• max — a legnagyobb fels̋o határú; a legbaloldalibb;

• ff — („first-fail” elv): a legkisebb tartományú (vö.fd_size ); a
legbaloldalibb;

• ffc — a legkisebb tartományú; a legtöbb korlátban előforduló (vö.
fd_degree ); a legbaloldalibb;

• variable(Sel) — (meta-opció)Sel egy felhasználói eljárás, amely
kiválasztja a következ̋o címkézend̋o változót (lásd 87. oldal).

A választás fajtája
A kiválasztottX változó tartományát a következőképpen bonthatjuk fel:

• step (alapértelmezés) —X #= B ésX #\= B közötti választás, aholB azX
tartományának alsó vagy felső határa (a bejárási iránytól függően);

• enum— többszörös választásX lehetséges értekei közül;

• bisect — X #=< MésX #> Mközötti választás, aholMazX
tartományának középső eleme (M= (min(X) + max(X))//2);

• value(Enum) — (meta-opció)Enumegy eljárás, amelynek az a feladata,
hogy leszűkítseX tartományát (lásd 88. oldal).

A bejárási irány
A tartomány bejárási iránya lehet:

• up (alapértelmezés) — alulról felfelé;

• down — felülről lefelé.

Címkézési opciók (folyt.)

A keresett megoldások

• all (alapértelmezés) — visszalépéssel az összes megoldást felsorolja;

• minimize(X) ill. maximize(X) — egy, azX-re minimális ill. maximális
értéket eredményező megoldást keres, branch-and-bound algoritmussal.

Példa széls̋oérték keresésére

| ?- _L=[X,Y,Z], domain(_L, 0, 1),
V#=Y+Z-X, labeling([minimize(V)], _L).

V = -1, X = 1, Y = 0, Z = 0 ? ;
no

A keresési fa a branch-and-bound algoritmussal

V#<0

V #<0

X=0

Y=0

Z=0

V=0

V= −1

V#<0

V= −1
X=1
Y=0
Z=0

Címkézési opciók (folyt.)

Egyéb opciók

• Statisztika:assumptions(K) — egyesítiK-t a sikeres megoldáshoz vezető
ágon lev̋o változó-kiválasztások számával (ami lényegében a keresési fában a
megoldáshoz vezető út hossza).

• A heurisztikától való eltérés korlátozása:discrepancy(D) (Dadott szám)—
csak olyan megoldásokat kérünk figyelembe venni, amelyekhez a keresési fában
úgy jutunk el, hogy a legfeljebbD-szer választunk egy nem-legbaloldalibb ágat
a választási pontokban. (Szemléletesen: a fa gyökerétől a megoldásig haladva
legfeljebbbD-szer kell megadni a jobbkéz-szabály szerinti elsőbbséget.)

Az opció háttere az LDS (Limited Discrepancy Search) keresési módszer.
Ebben feltételezzük, hogy a legbaloldalibb választások képviselik azt a
heurisztikát, amivel nagy valószínűséggel eljuthatunk egy megoldáshoz. Mivel
a heurisztika nem teljesen tökéletes, ezért valamennyi eltérést megengedünk, de
az össz-eltérés-mennyiséget korlátozzuk.

• időkorlát: time_out(MSec,Result) . HaMSecmilliszekundum alatt
lefut, Result = success , egyébként lelövi a cimkézést ésResult =
time_out . A minimize/maximize opciókkal jól működik együtt (ezek az
opciók az addigi legjobb eredményt adják vissza).

Példák (vö. a 81. oldalon lev̋o keresési fákkal):

assumptions(Select, As) :-
X in 1..4,
findall(A, labeling([Select, assumptions(A)], [X]), As) .

lds(Select, D, Xs) :-
X in 1..4,
findall(X, labeling([Select, discrepancy(D)], [X]), Xs) .

| ?- assumptions(enum, As). As = [1,1,1,1]
| ?- assumptions(bisect, As). As = [2,2,2,2]
| ?- assumptions(step, As). As = [1,2,3,3]

| ?- lds(enum, 1, Xs). Xs = [1,2,3,4]
| ?- lds(bisect, 1, Xs). Xs = [1,2,3]
| ?- lds(step, 1, Xs). Xs = [1,2]

A címkézés testreszabása

labeling/2 — a variable(Sel) meta-opció

• variable(Sel) — Sel egy eljárás, amely kiválasztja a következő
címkézend̋o változót.Sel(Vars,Selected,Rest) alakban hívja meg a
rendszer, aholVars a még címkézend̋o változók/számok listája.

• Sel -nek determinisztikusan sikerülnie kell egyesítveSelected -et a
címkézend̋o változóvalésRest -et a maradékkal.

• Sel egy tetsz̋oleges meghívható kifejezés lehet (callable , azaz név vagy
struktúra). A három argumentumot a rendszer fűziSel argumentumlistájának
végére.

• Például: ha aSel opcióként amod:sel(Param) kifejezést adjuk meg, akkor
a rendszer amod:sel(Param,Vars,Selected,Rest) eljáráshívást
hajtja majd végre.

Példa avariable opció használatára

% A Vars-beli változók között Sel a Hol-adik,
% Rest a maradék.
valaszt(Hol, Vars, Sel, Rest) :-

szur(Vars, Szurtek),
length(Szurtek, Len), N is integer(Hol * Len),
nth0(N, Szurtek, Sel, Rest).

% szur(Vk, Szk): A Vk-ban lev ő változók listája Szk.
szur([], []).
szur([V|Vk], Szk) :- nonvar(V), !, szur(Vk, Szk).
szur([V|Vk], [V|Szk]) :- szur(Vk, Szk).

queens([], 8, Qs). → Qs = [1,5,8,6,3,7,2,4]
queens([variable(valaszt(0.5))], 8, Qs)

→ Qs = [7,2,6,3,1,4,8,5]
queens([variable(valaszt(0.7))], 8, Qs)

→ Qs = [5,7,2,6,3,1,4,8]

A címkézés testreszabása (folyt.)

labeling/2 — a value(Enum) meta-opció

• value(Enum) — Enumegy eljárás, amelynek az a feladata, hogy leszűkítse
X tartományát. Az eljárást a rendszerEnum(X, Rest, BB0, BB) alakban
hívja meg, ahol[X|Rest] a még címkézend̋o változók listája.

• Enum-nak nemdeterminisztikusan le kell szűkítenieX tartományát az összes
lehetséges módon, vö. a címkézés menetének leírását a 82. oldalon. (Avalue
opció ac., d. ése. lépések együttesét váltja ki.)

• Az els̋o választásnál meg kell hívnia azfirst_bound(BB0, BB) , a
kés̋obbieknél alater_bound(BB0, BB) eljárást, a BB ill. LDS keresési
algoritmusok kiszolgálására.

• Enum-nak egy meghívható kifejezésnek kell lennie. A négy argumentumot a
rendszer fűziEnumargumentumlistájának a végére.

Példa: belülről kifelé való érték-felsorolás

midout(X, _Rest, BB0, BB) :-
fd_size(X, Size),
Mid is (Size+1)//2,
fd_set(X, Set),
fdset_to_list(Set, L),
nth1(Mid, L, MidElem),
( first_bound(BB0, BB), X = MidElem
; later_bound(BB0, BB), X #\= MidElem
).

| ?- X in {1,3,12,19,120},
labeling([value(midout)], [X]).

X = 12 ? ;
X = 3 ? ;
X = 19 ? ;
X = 1 ? ;
X = 120 ? ; no

88



A címkézés hatékonysága

A korábbiqueens eljárás megoldásai 600 MHz Pentium III gépen.

Összes megoldás keresése

méret n=8 n=10 n=12

megoldások száma 92 724 14200

címkézés sec btrk sec btrk sec btrk

[step] 0.07 324 1.06 5942 25.39 131K

[enum] 0.07 324 1.03 5942 24.84 131K

[bisect] 0.07 324 1.07 5942 26.04 131K

[enum,min] 0.08 462 1.31 8397 33.89 202K

[enum,max] 0.07 462 1.31 8397 33.89 202K

[enum,ff] 0.06 292 0.97 4992 21.57 101K

[enum,ffc] 0.06 292 1.04 4992 23.24 101K

[enum, midvar 1] 2 0.06 286 0.90 4560 20.11 88K

Első megoldás keresése

méret n=16 n=18 n=20

címkézés sec btrk sec btrk sec btrk

[enum] 0.43 1833 1.76 7436 9.01 37320

[enum,min] 0.52 2095 0.87 2595 1.39 3559

[enum,max] 0.61 3182 2.68 13917 16.06 83374

[enum,ff] 0.03 7 0.05 11 0.08 33

[enum,ffc] 0.03 7 0.05 11 0.09 33

[enum, midvar 1] 2 0.04 69 0.06 57 0.15 461

[value(midout) 2] 0.04 3 0.05 4 0.09 38

[value(midout) 2,ffc] 0.04 15 0.06 41 0.08 20

1midvar ≡ variable(valaszt(0.5)) .
2Hatékonyabb statikusan (a címkézés előtt egyszer) elrendezni a változókat és az értékeket,

lásd azalt_queens/2 eljárást alibrary(’clpfd/examples/queens’) állományban.

Széls̋oértékek ismételt hívással való előállítása

minimize(Cél, X) ill. maximize(Cél, X)
A Cél ismételt hívásávalmegkeresi azX változó minimális ill. maximális értékét.

A minimize/2 eljárás definíciója

my_minimize(Goal, Var) :-
findall(Goal-Var, (Goal -> true), [Best1-UB1]),
minimize(Goal, Var, Best1, UB1).

% minimize(Goal, Var, BestSoFar, UB): Var is the minimal val ue < UB
% allowed by Goal, or, failing that, Goal = BestSoFar and Var = UB.
minimize(Goal, Var, _, UB) :- var(UB), !, error.

% Goal does not instantiate Var
minimize(Goal, Var, _, UB) :-

call( Var #< UB ) , % csak a lenti nyomkövetés kedvéért
findall(Goal-Var, (Goal -> true), [Best1-UB1]), !,
minimize(Goal, Var, Best1, UB1).

minimize(Goal, Var, Goal, Var).

Magyarázatok a fenti definícióhoz

• findall(Cél, (Cél->true), [EM]) : EMa Cél els̋o megoldásának másolata.

• A keresési fa szerkezetétől függ, hogy aminimize/2 vagy a
labeling([minimize...],...) a hatékonyabb. Pl. aminimize/2 a 85.
oldalon lev̋o fában elkerüli azX,Y -hoz tartozó választási pontok bejárását.

Példa amy_minimize/2 használatára

p(L, V) :- L = [X,Y,Z], domain(L, 0, 1), V #= Y+Z-X.

| ?- spy [call/1,minimize/4,labeling/2].
| ?- p(L, V), my_minimize(labeling([], L), V).

+ 1 1 Call: lblg (user:[],[ X,Y,Z ]) ? z
?+ 1 1 Exit: lblg (user:[],[0,0,0]) ? z

+ 2 1 Call: minimize( lblg ([],[ X,Y,Z ]), V, lblg ([],[0,0,0]),0) ? z
+ 3 2 Call: call(user:( V#<0)) ? z
+ 3 2 Exit: call(user:(-1#<0)) ? z
+ 4 2 Call: lblg (user:[],[1,0,0]) ? z
+ 4 2 Exit: lblg (user:[],[1,0,0]) ? z
+ 5 2 Call: minimize( lblg ([],[1,0,0]),-1, lblg ([],[1,0,0]),-1) ? z
+ 6 3 Call: call(user:(-1#< -1)) ? z
+ 6 3 Fail: call(user:(-1#< -1)) ? z
+ 5 2 Exit: minimize( lblg ([],[1,0,0]),-1, lblg ([],[1,0,0]),-1) ? z
+ 2 1 Exit: minimize( lblg ([],[1,0,0]),-1, lblg ([],[0,0,0]),0) ? z

L = [1,0,0], V = -1 ?

2. kis házi feladat: számkeresztrejtvény

A feladat

• Adott egy keresztrejtvény, amelyek egyes kockáiba1..Maxszámokat kell
elhelyezni (szokásosanMax= 9).

• A vízszintes és függ̋oleges „szavak” meghatározásaként a benne levő számok
összege van megadva.

• Egy szóban lev̋o betűk (kockák) mind különböző értékkel kell bírjanak.

A keresztrejtvény Prolog ábrázolása:

• listák listájaként megadott mátrix;

• a fekete kockák helyénF \V alakú struktúrák vannak, aholF ésV az adott
kockát követ̋o függ̋oleges ill. vízszintes szó összege, vagyx , ha nincs ott szó,
vagy egy egybetűs szó van;

• a kitöltend̋o fehér kockákat (különböző) változók jelzik.

A megirandó Prolog eljárás és használata

% szamker(SzK, Max): SzK az 1..Max számokkal
% helyesen kitöltött számkeresztrejtvény.
% Megjegyzés: egyes sorban/oszlopban középen
% is lehet ‘x’!

pelda(mini, [[x\ x,11\x,21\x, 8\x],
[x\24, _, _, _],
[x\10, _, _, _],
[x\6, _, _, x\x]], 9).

| ?- pelda(mini, SzK, _Max), szamker(SzK, _Max).
SzK = [[x\x, 11\x,21\x,8\x],

[x\24,8, 9, 7 ],
[x\10,2, 7, 1 ],
[x\6, 1, 5, x\x]] ? ; no

11 8

10

6

24
21

2

1

8 9 7

7 1

5

Kombinatorikus (szimbolikus) korlátok

A kombinatorikus korlátok általános tulajdonságai

• A korlátok nem tükrözhetőek.

• Az argumentumaikban szereplő FD változók helyett mindig írható egész szám.

Értékek számolása
count(Val, List, Relop, Count)
Jelentése: aVal egész szám aList FD-változó-listábann-szer fordul el̋o, és
fennáll az „n Relop Count ” reláció. Itt Count FD változó,Relop a hat
összehasonlító reláció egyike:#=, #\=, #< . . . . Tartomány-szűkítést biztosít.

global_cardinality(Vars, Vals)
Vars egy FD változókból álló lista,Vals pedigI-K alakú párokból álló lista, ahol
I egy egész,K pedig egy FD változó. MindegyikI érték csak egyszer fordulhat elő
aVals listában. Jelentése: AVars -beli FD változók csak a megadottI értékeket
vehetik fel, és minden egyesI-K párra igaz, hogy aVars listában pontosanK
darabI értékű elem van. Tartomány-szűkítést ad, haVals vagyVars tömör, és
még sok más speciális esetben.

Példa: mágikus sorozatok, újabb változatok
% Az L lista egy N hosszúságú mágikus sorozatot ír le.
magikus(N, L) :-

length(L, N), N1 is N-1, domain(L, 0, N1),

eloford(L, 0,
L, Egyhat),

parok(L, 0, Pk, Egyhat),
global_cardinality(L, Pk),

sum(L, #=, N), scalar_product(Egyhat, L, #=, N),
labeling([], L).

% eloford([E i, E i+1, ...], i, Sor, Egyhat):
% Sor-ban az i szám Ei-szer, az i + 1 szám Ei+1-szer stb.
% fordul el ő. Egyhat az [ i, (i + 1),...] együttható-lista.
eloford([], _, _, []).
eloford([E|Ek], I, Sor, [I|EH]) :-

count(I, Sor, #=, E),
J is I+1, eloford(Ek, J, Sor, EH).

% parok([E i, E i+1, ...], i, Parok, Egyhat):
% Parok az [ i-E i, (i + 1)-E i+1, ...] párlista,
% Egyhat az [ i, (i + 1),...] együttható-lista.
parok([], _, [], []).
parok([E|Ek], I, [I-E|Pk], [I|EH]) :-

J is I+1, parok(Ek, J, Pk, EH).

